

Petunjuk Praktikum  
(Edisi Revisi)

**Pemrograman Komputer dengan Turbo Pascal  
versi  
Windows 1.5**



Drs. Sahid, MSc.



# Daftar Isi

<b>Kata Pengantar</b> .....	<b>iii</b>
<b>Kata Pengantar Edisi Revisi</b> .....	<b>iv</b>
<b>Praktikum 1 (Pengenalan Turbo Pascal for Windows dan Struktur program Pascal)</b> .....	<b>1</b>
A. Peralatan yang diperlukan: .....	1
B. Tujuan Praktikum .....	1
C. Dasar Teori .....	1
C. Kegiatan Praktek .....	4
E. Laporan Praktikum .....	6
<b>Praktikum 2 (Pengenalan variabel, Operasi Aritmetika, dan Input-Output)</b> .....	<b>9</b>
A. Peralatan yang diperlukan: .....	9
B. Tujuan Praktikum .....	9
C. Dasar Teori .....	9
D. Kegiatan Praktek .....	14
E. Laporan Praktikum .....	16
<b>Praktikum 3 (Label, Goto, Operator Logika, IF, dan CASE)</b> .....	<b>19</b>
A. Peralatan yang diperlukan: .....	19
B. Tujuan Praktikum .....	19
C. Dasar Teori .....	19
D. Kegiatan Praktek .....	21
E. Laporan Praktikum .....	24
<b>Praktikum 4 (Pengulangan Proses: <i>Looping</i> dengan FOR ... TO ... DO, WHILE ... DO, ... REPEAT ... UNTIL ...)</b> .....	<b>27</b>
A. Peralatan yang diperlukan: .....	27
B. Tujuan Praktikum .....	27
C. Dasar Teori .....	27
D. Kegiatan Praktek .....	28
E. Laporan Praktikum .....	31
<b>Praktikum 5 (Penggunaan Fungsi-fungsi Matematika)</b> .....	<b>32</b>
A. Peralatan yang diperlukan: .....	32
B. Tujuan Praktikum .....	32
C. Dasar Teori .....	32
D. Kegiatan Praktek .....	33
E. Laporan Praktikum .....	36
<b>Praktikum 6 (Enumerasi dan Himpunan)</b> .....	<b>37</b>
A. Peralatan yang diperlukan: .....	37

B. Tujuan Praktikum.....	37
C. Dasar Teori.....	37
D. Kegiatan Praktek.....	39
E. Laporan Praktikum.....	42
<b>Praktikum 7 (Array: Variabel Berdimensi / Jajaran Variabel) .....</b>	<b>43</b>
A. Peralatan yang diperlukan:.....	43
B. Tujuan Praktikum.....	43
C. Dasar Teori.....	43
D. Kegiatan Praktek.....	44
E. Laporan Praktikum.....	48
<b>Praktikum 8 (Variabel String: Operasi dan Fungsi-fungsi String) .....</b>	<b>49</b>
A. Peralatan yang diperlukan:.....	49
B. Tujuan Praktikum.....	49
C. Dasar Teori.....	49
D. Kegiatan Praktek.....	50
E. Laporan Praktikum.....	54
<b>Praktikum 9 (Pemrograman Prosedural: Prosedur, Fungsi, dan Rekursi).....</b>	<b>55</b>
A. Peralatan yang diperlukan:.....	55
B. Tujuan Praktikum.....	55
C. Dasar Teori.....	55
D. Kegiatan Praktek.....	56
E. Laporan Praktikum.....	64
<b>Praktikum 10 (Bekerja dengan RECORD).....</b>	<b>66</b>
A. Peralatan yang diperlukan:.....	66
B. Tujuan Praktikum.....	66
C. Dasar Teori.....	66
D. Kegiatan Praktek.....	67
E. Laporan Praktikum.....	72
<b>Praktikum 11 (Bekerja dengan File: Membaca dan Menulis Data dari / ke File) .....</b>	<b>74</b>
A. Peralatan yang diperlukan:.....	74
B. Tujuan Praktikum.....	74
C. Dasar Teori.....	74
D. Kegiatan Praktek.....	75
E. Laporan Praktikum.....	80
<b>Daftar Pustaka .....</b>	<b>82</b>

# Kata Pengantar

Puji syukur Alhamdulillah penulis panjatkan ke Hadhirat Allah SwT. atas nikmat kesehatan dan kekuatan yang diberikan kepada penulis, sehingga penulisan petunjuk praktikum ini dapat diselesaikan.

Petunjuk Praktikum ini disusun untuk dapat digunakan sebagai panduan pelaksanaan kegiatan praktikum pada mata kuliah **Pemrograman Komputer** pada program studi Matematika dan Pendidikan Matematika FMIPA Universitas Negeri Yogyakarta.

Petunjuk Praktikum ini terdiri atas 11 modul praktikum, meliputi: (1) *Pengenalan Turbo Pascal for Windows dan Struktur program Pascal*, (2) *Pengenalan variabel, Operasi Aritmetika, dan Input-Output*, (3) *Label, Goto, Operator Logika, IF, dan CASE*, (4) *Pengulangan Proses: Looping dengan FOR ... TO ... DO, WHILE ... DO, ... REPEAT ... UNTIL ...*, (5) *Penggunaan Fungsi-fungsi Matematika*, (6) *Enumerasi dan Himpunan*, (7) *Array: Variabel Berdimensi / Jajaran Variabel*, (8) *Variabel String: Operasi dan Fungsi-fungsi String*, (9) *Pemrograman Prosedural: Prosedur, Fungsi, dan Rekursi*, (10) *Bekerja dengan RECORD*, dan (11) *Bekerja dengan File: Membaca dan Menulis Data dari / ke File*. Materi-materi praktikum tersebut mendukung kegiatan pembelajaran mata kuliah Pemrograman Komputer satu semester dan meliputi 4 SKS (2 SKS Teori dan 2 SKS Praktek).

Penulis menyadari bahwa tiada gading yang tak retak. Saran dan masukan dari berbagai pihak, baik dosen maupun mahasiswa dengan senang hati akan penulis terima sebagai bahan untuk pengembangan dan perbaikan petunjuk praktikum ini. Penulis menyampaikan banyak terima kasih kepada berbagai pihak yang telah membantu penyusunan petunjuk praktikum ini.

Yogyakarta, 26 Nopember 2002

Penulis

# Kata Pengantar Edisi Revisi

Alhamdulillah, puji syukur penulis panjatkan ke Hadhirat Allah SwT atas nikmat kesehatan dan kekuatan yang diberikan kepada penulis, sehingga revisi petunjuk praktikum ini dapat diselesaikan.

Revisi ini dimaksudkan sebagai penyempurnaan petunjuk praktikum Pemrograman Komputer. Materi praktikum tidak mengalami perubahan, masih sama dengan edisi sebelumnya. Meskipun demikian penambahan **Dasar Teori** untuk setiap kegiatan praktikum telah ditambahkan pada edisi revisi ini, selain penyempurnaan eitorial. Diharapkan dengan adalah bagian dasar teori tersebut mahasiswa di dalam kegiatan praktikum pemrograman dengan Turbo Pascal menjadi lebih memahami apa yang dilakukan.

## Catatan untuk mahasiswa:

Di dalam melakukan kegiatan praktikum menyusun program Pascal, sebaiknya mahasiswa untuk pertama kalinya mengikuti contoh-contoh program yang ada dan berusaha melakukan **debuging**, yakni membetulkan program, baik dari segi **Syntax Error** maupun **logical error**. Syntax error lebih mudah dibetulkan, karena Pascal tidak akan mau menjalankan program selagi program masih salah. Logical error dapat dihindari dengan menguji program menggunakan berbagai kemungkinan masukan dan mengecek hasil (outputnya), apakah sudah benar. Apabila hasilnya salah berarti programnya salah, harus dibetulkan. Dengan demikian Anda tidak hanya belajar mengetik program, tetapi benar-benar belajar menjadi seorang programmer. Apabila perlu Anda dapat memodifikasi contoh program yang ada untuk melakukan hal-hal yang Anda inginkan.

Akhirnya, penulis berharap petunjuk praktikum ini dapat bermanfaat.

Yogyakarta, 12 Februari 2004

Sahid

# Praktikum 1

## (Pengenalan Turbo Pascal for Windows dan Struktur program Pascal)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menjalankan program Turbo Pascal for Windows.
2. Mahasiswa dapat menyebutkan bagian-bagian layar Turbo Pascal for Windows dan kegunaannya.
3. Mahasiswa dapat menyebutkan daftar menu dan sub-submenu pada Turbo Pascal for Windows.
4. Mahasiswa dapat menyebutkan kegunaan menu dan sub-submenu pada Turbo Pascal for Windows.
5. Mahasiswa dapat menggunakan tombol-tombol untuk menyunting program Pascal.
6. Mahasiswa dapat menulis program Pascal sederhana yang menggunakan perintah `writeln` dan `writeln` secara benar.
7. Mahasiswa dapat menyebutkan struktur dan tata tulis program Pascal.

### C. Dasar Teori

Struktur suatu program Pascal terdiri atas bagian-bagian sebagai berikut.

Bagian	Pola	Contoh
Judul Program	<b>Program</b> NamaProgram;	Program HasilKaliJumlah;
	<b>Uses</b> Unit1, Unit2, ...;	Uses WinCrt;
Deklarasi	<b>Var</b> Var1:jenis; Var2,Var3, ... : jenis; .... <b>Const</b> Const1=nilai; Const2=nilai; .....	Var a,b,c,d:real; m,n,s:real;
	<b>Type</b> Pengenal1 = NamaType; ... <b>Label</b> label1,label2, ...;	Begin Read(a); read(b); read(c); read(d);
	<b>Procedure</b> NamaProsedur; <b>begin</b> .... {definisi prosedur} <b>end</b> ;	m:=a+b; n:=c+d; s:=m*n;
	<b>Function</b> NamaFungsi; <b>begin</b> .... {definisi fungsi} <b>end</b> ;	writeln(s); End.
Bagian Utama (Tubuh Program)	<b>Begin</b> Statement1; Statement2; .... <b>End.</b>	

Bagian Judul program berguna untuk memberi nama program dan parameter. Bagian ini sifatnya

dekoratif (untuk menambah keterbacaan program) dan tidak ada artinya sama sekali bagi program itu sendiri.

Kata **uses** digunakan untuk memanggil nama-nama **Unit** yang akan digunakan di dalam program. Bagian Deklarasi digunakan untuk mendefinisikan nama-nama pengenalan (*Identifier*) baru, seperti nama variabel, konstanta, label, jenis data baru (*type*), nama fungsi, dan nama prosedur. Definisi nama-nama pengenalan baru dapat diulangi berkali-kali dan dengan urutan sebarang.

Kata **label** digunakan untuk mendefinisikan nama-nama label yang menandai beberapa statement di dalam bagian utama program Pascal. Setiap label hanya menandai satu statement. Nama label harus unik, tidak boleh sama dengan nama label atau pengenalan yang sudah ada. Nama label dapat menggunakan abjad maupun angka (atau kombinasi abjad dan angka 0 ... 9999). Di dalam program, nama label ditulis di depan suatu statement dan diakhiri dengan titik dua (:). Untuk meloncat ke suatu statement yang berlabel digunakan perintah **goto NamaLabel**

#### Contoh:

```
label 1, 2;
goto 1
....
1:  x1:= (-b+sqrt(D))/(2*a);
    x2:= (-b-sqrt(D))/(2*a);
2:  x := -b/(2*a);
```

Kata **Const** digunakan untuk mendefinisikan nama-nama konstanta dan nilainya.

```
const
  NamaKonstanta1 = nilai;
  NamaKonstanta2 = nilai;
  ...
```

Nilai yang disimpan dalam suatu konstanta dapat berupa bilangan atau nilai dari suatu ekspresi yang memuat operasi-operasi aritmetika atau perhitungan suatu fungsi. Syaratnya ekspresi tersebut harus dapat dihitung nilainya.

#### Contoh:

```
(* Deklarasi beberapa konstanta *)
const
  MaxData   = 1024 * 64 - 16;
  CacahAbjad = Ord('Z') - Ord('A') + 1;
  Pesan     = 'Selamat Datang ...';
```

Notasi desimal biasa digunakan untuk menyatakan konstanta bilangan (data) jenis integer dan real. Konstanta integer heksadesimal ditulis dengan menggunakan tanda dolar (\$) sebagai awalnya. Untuk menyatakan konstanta real dapat digunakan notasi ilmiah (dengan huruf **E** atau **e** diikuti nilai pangkat) dan dibaca "kali 10 pangkat ...". Misalnya,

7E-2 berarti  $7 \times 10^{(-2)}$   
12.25e+6 atau 12.25e6 berarti  $12.25 \times 10^{(+6)}$ .

Semua bilangan yang memiliki titik desimal atau pangkat adalah konstanta berjenis **real**. Setiap bilangan desimal lain merupakan jenis integer, nilainya harus dalam jangkauan dari -2,147,483,648 sampai 2,147,483,647. Bilangan-bilangan heksadesimal merupakan konstanta jenis integer yang memiliki jangkauan nilai dari \$00000000 sampai \$FFFFFFFF.



Kata **Const** juga dapat digunakan untuk mendefinisikan suatu variabel dan memberi nilai awal pada variabel tersebut. Formatnya adalah sebagai berikut:

```
const
  Var1: Jenis = Nilai;
  var2: jenis = Nilai;
  ...
```

Kata **Var** digunakan untuk mendefinisikan variabel-variabel dan jenisnya. Setiap jenis variabel akan disimpan di dalam memori dengan lokasi yang berbeda. Suatu variabel akan digunakan untuk menyimpan nilai paling akhir yang diberikan kepadanya. Format pendefinisian variabel adalah sebagai berikut:

```
var
  Var1, ..., VarN: jenis;
  NamaVar: jenis;
  ...
```

Kata **absolute** dapat digunakan untuk memesan alamat mutlak di dalam memori komputer. Kata **var** juga dapat digunakan untuk mendeklarasi jenis parameter di dalam suatu fungsi atau prosedur.

#### Contoh:

```
{ Deklarasi beberapa variabel}
var
  X, Y, Z: real;
  I, J, K: Integer;
  Selesai, Salah: Boolean;
  Vektor: array[1..10] of real;
  Nama: string[15];
  BerkasI, BerkasO: Text;
  Abjad: set of 'A'..'Z';
```

Kata **type** digunakan untuk mendeklarasikan suatu pengenal (identifier) yang menunjukkan suatu [type](#). Format deklarasinya adalah sebagai berikut:

```
type
  Jenis1 = NamaType;
  jenis2 = NamaType;
  ...
```

NamaType adalah salah satu dari: **array**, **file**, **object**, **ordinal**, **pointer**, **real**, **record**, **set**, **string**.

Pada bagian deklarasi juga dapat didefinisikan suatu prosedur atau fungsi. Suatu prosedur adalah bagian dari suatu program yang digunakan untuk melakukan suatu tugas tertentu, mungkin dengan menggunakan beberapa parameter tertentu.

Bagian utama suatu program Pascal diawali dengan kata **Begin** dan diakhir dengan kata **End**. (ada titiknya!) Antara **Begin** dan **End**. dapat ditulis beberapa statemen sesuai dengan kegunaan program tersebut.

Perintah-perintah berikut berguna untuk mengontrol jalannya program.


Halt	Halt [ ( Exitcode: Word ) ];	Menghentikan jalannya program dan kembali ke
------	------------------------------	--


		sistem Windows/DOS.
Exit	Exit;	Keluar dari blok program (loop, prosedur, fungsi, atau program).
RunError	RunError [ ( Errorcode: Byte ) ];	Menghentikan jalannya program.

### C. Kegiatan Praktek

Pada praktikum pertama ini, baca secara cermat dan praktekan petunjuk-petunjuk di bawah ini, sehingga Anda benar-benar paham dan familier cara pemakaian Turbo Pascal for Windows. Apabila perlu hafalkan pemakaian tombol-tombol dan menu-menu yang ada. Hal ini akan memudahkan dan mempercepat Anda dalam menyunting program.

#### 1. Menjalankan program Turbo Pascal for Windows 1.5:

Klik ikon  pada layar dekstop atau klik menu **Start** → **Programs** → **TPW 1.5** → **TPW 1.5**, sehingga muncul layar Turbo Pascal for Windows seperti gambar di bawah ini.



Gambar 1:Layar program Turbo Pascal for Windows Versi 1.5

#### 2. Mengetal Layar Turbo Pascal for Windows 1.5:

Gambar 1 menampilkan layar program Turbo Pascal for Windows versi 1.5. Layar tersebut merupakan sebuah IDE (*Integrated Development Environment*), yang terdiri atas layar utama Turbo Pascal dan layar editor (tempat untuk mengedit program). Seperti pada gambar tersebut, layar Turbo Pascal versi Windows 1.5 terdiri atas bagian-bagian sebagai berikut:

- 1) **Logo dan simbol program Turbo Pascal**
- 2) **Baris menu:** setiap menu merupakan kumpulan perintah Turbo Pascal
- 3) **Baris batang tool (toolbar),** yang merupakan simbol perintah-perintah Turbo Pascal
- 4) **Baris status:** posisi kursor pada layar editor, modus tindih (*Insert*) / geser (*Overwrite*), dll.
- 5) **Tombol untuk mengecilkan, memperbesar, dan menutup layar**
- 6) **Layar editor:** tempat menuliskan program Pascal
- 7) **Batang penggeser layar (scrollbar),** untuk menggeser dan menggulung layar editor.

#### 3. Mengetal lebih jauh tentang menu-menu Turbo Pascal:

- 1) Klik menu **File**, sebutkan sub-submenu yang ada di bawahnya.

- 2) Klik menu **Edit**, sebutkan sub-submenu yang ada di bawahnya.
- 3) Klik menu **Search**, sebutkan sub-submenu yang ada di bawahnya.
- 4) Klik menu **Run**, sebutkan sub-submenu yang ada di bawahnya.
- 5) Klik menu **Compile**, sebutkan sub-submenu yang ada di bawahnya.
- 6) Klik menu **Options**, sebutkan sub-submenu yang ada di bawah nya.
- 7) Klik menu **Window**, sebutkan sub-submenu yang ada di bawahnya.
- 8) Klik menu **Help**, sebutkan sub-submenu yang ada di bawahnya.

4. Latihan menulis program

Di dalam menyunting atau menulis program dengan editor Turbo Pascal, Anda dapat menggunakan tombol-tombol sebagai berikut untuk menggerakkan kursor.

Gerakan kursor ke ..	Tombol yang dipakai	Alternatif lain
Satu karakter ke kiri	Panah kiri (←)	Ctrl+S
Satu karakter ke kanan	Ctrl+[→]	Ctrl+D
Satu kata ke kiri	Ctrl+[←]	Ctrl+A
Gulung ke bawah satu baris	Ctrl+Z	Ctrl+Z
Satu baris ke atas	Up	Ctrl+E
Satu baris ke bawah	Down	Ctrl+X
Gulung ke atas satu baris	Ctrl+W	Ctrl+W
Satu layar ke atas	PgUp	Ctrl+R
Satu layar ke bawah	PgDn	Ctrl+C
Posisi kursor terakhir	-	Ctrl+Q P
Awalbaris	Home	Ctrl+Q S
Akhir baris	End	Ctrl+Q D
Bagian atas layar	Ctrl+E	Ctrl+Q E
Bagian bawah layar	Ctrl+X	Ctrl+Q X
Awal berkas	Ctrl+Home	Ctrl+Q R, Ctrl+PgUp
Akhir berkas	Ctrl+End	Ctrl+Q C, Ctrl+PgDn

Untuk menghapus dan menyisipkan teks dapat digunakan tombol-tombol sebagai berikut

Menghapus/menyisipkan	Tombol yang dipakai	Alternatif lain
Satu karakter di kanan kursor	Del	Ctrl+G
Satu karakter di kiri kursor	Backspace	-
Menghapus baris	Ctrl+Y	-
Menghapus sampai akhir baris	Shift+Ctrl+Y	Ctrl+Q Y
Menghapus kata	-	Ctrl+T
Menyisipkan baris	Ctrl+N	Ctrl+N
Mengganti modus sisip/tindih	Ins	Ctrl+V

Selain menggunakan tombol, Anda dapat menggunakan mouse untuk memindahkan kursor. Arahkan penunjuk pada posisi yang diinginkan, lalu klik. Anda juga dapat memblok tulisan dengan cara mengklik awal blok, lalu tekan tombol mouse kiri sambil digeser ke akhir blok. Tulisan yang diblok dapat dihapus dengan menekan tombol **Del** atau mengklik menu **Edit → Cut**.

Pada saat pertama kali dijalankan, Turbo Pascal secara otomatis (sebelum diubah) akan membuka berkas contoh program bernama `welcome.pas`, seperti terlihat pada Gambar 1. Pada Turbo Pascal, **tulisan yang diapit oleh tanda kurung kurawal { }** merupakan **komentar atau penjelasan program, tidak akan dikerjakan oleh komputer**. Komentar boleh tidak ditulis, namun keberadaannya membantu pembaca program.

Apabila layar editor Anda berisi program lain, hapus program tersebut, tulis program `welcome.pas` tersebut. Demikian pula jika layar editor Anda kosong, tulis program tersebut. Untuk menulis program baru klik menu **File** → **New**.

- 2) Dengan menggunakan program **Windows Explorer**, buat direktori baru untuk menyimpan program-program Anda, misalnya `C:\tpw\mat0015` (Mahasiswa prodi Mat angkatan 2000 dengan nomor urut 15). **Ingat, jangan menyimpan program Pascal dengan atau di direktori dengan nama lebih dari 8 karakter atau tulisan dipisah, misalnya coba `1.pas`, `My Document\cobal.pas`. TPW adalah program yang tidak bisa bekerja dengan nama file panjang.**
- 3) Simpan kembali program `welcome.pas` dengan nama `selamat.pas` di direktori Anda: **Klik File** → **Save As** → (Pilih direktori, misalnya `C:\tpw\mat0015`) → (Tulis nama file `selamat.pas` → **klik Ok**).
- 4) Compile program Anda: Klik menu **Compile** → **Compile**
- 5) Jalankan program Anda dan amati apa yang terjadi: Klik menu **Run** → **Run**
- 6) Ubah program Anda sehingga jika dijalankan akan menampilkan tulisan dengan format:

**Selamat Datang**  
**Anda sudah dapat membuat program baru**  
**dengan menggunakan Turbo Pascal versi Windows 1.5**

- 7) Simpan kembali program Anda: Klik menu **File** → **Save** (Anda tidak lagi perlu memberi nama!)

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

A. Judul Praktikum : **Praktikum Pemrograman Komputer**

B. Identitas Praktikan :  
 Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_  
 Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_

C. Hasil Praktikum :

1. Buat dan lengkapi tabel-tabel seperti di bawah ini
  - a. Daftar perintah (submenu) pada menu **File**:

Perintah (submenu)	Kegunaan	Tombol jalan pintas (short-cut) (jika ada)

- b. Daftar perintah (submenu) pada menu **Edit**:

Perintah (submenu)	Kegunaan	Tombol jalan pintas (short-cut) (jika ada)

- c. Daftar perintah (submenu) pada menu **Search**:

Perintah (submenu)	Kegunaan	Tombol jalan pintas (short-cut) (jika ada)

- d. Daftar perintah (submenu) pada menu **Run**:

Perintah (submenu)	Kegunaan	Tombol jalan pintas (short-cut) (jika ada)

e. Daftar perintah (submenu) pada menu **Compile**:

Perintah (submenu)	Kegunaan	Tombol jalan pintas (short-cut) (jika ada)

f. Daftar perintah (submenu) pada menu **Option**:

Perintah (submenu)	Kegunaan	Tombol jalan pintas (short-cut) (jika ada)

g. Daftar perintah (submenu) pada menu **Window**:

Perintah (submenu)	Kegunaan	Tombol jalan pintas (short-cut) (jika ada)

h. Daftar perintah (submenu) pada menu **Help**:

Perintah (submenu)	Kegunaan	Tombol jalan pintas (short-cut) (jika ada)

2. Lakukan dan lengkapi tabel di bawah ini.

<b>Lakukan hal-hal sebagaimana berikut:</b>	Klik menu <b>Run → Run</b> . Tuliskan apa yang terjadi. Jelaskan mengapa demikian?
Ubah nama di belakang kata <b>program</b> pada baris pertama menjadi: Selamat (tanpa titik koma)	
Hapus kata <code>wincrt ;</code>	
Ubah baris-baris antara <b>begin ... end</b> dengan baris-baris sebagai berikut: <code>Selamat1:='Selamat Belajar Turbo Pascal';</code> <code>Selamat2:='Program Anda harus menggunakan tiga buah variabel';</code> <code>Selamat3:='berjenis string';</code> <code>Writeln(selamat1);</code> <code>Writeln(selamat2);</code> <code>Writeln(selamat3);</code> Simpan program Anda dengan nama <code>tutorial1.pas</code>	

Tambahkan perintah berikut sebelum baris <b>begin:</b> Var selamat1, selamat2, selamat3;	
Ganti perintah-perintah writeln di atas dengan perintah write.	
Ganti ketiga perintah write tersebut dengan baris berikut: Write(selamat1, #10#13, selamat2, #10#13, selamat3); Simpan program Anda dengan nama kontrol.pas	
Hapus tanda titik pada akhir program.	
Ganti semua perintah dengan huruf besar.	
Hapus baris pertama yang memuat kata program.	

3. Jawablah pertanyaan-pertanyaan di bawah ini.
- Apakah kegunaan perintah Uses?
  - Haruskah setiap program Pascal diawali dengan kata program?
  - Sebutkan aturan penulisan nama di belakang kata program!
  - Apakah kegunaan perintah var?
  - Apakah Pascal membedakan huruf besar dan huruf kecil?
  - Apakah kegunaan tanda titik pada program Pascal?
  - Apakah kegunaan tand titik koma pada program Pascal?
  - Apakah kegunaan tanda koma pada program Pascal?
  - Apakah kegunaan tanda petik tunggal ('...') pada program Pascal?
  - Apakah kegunaan tanda titik dua diikuti sama dengan (:=) pada program Pascal?
  - Sebutkan tombol-tombol singkat untuk melakukan:
    - mengkompile program
    - menjalankan program
    - menghapus tulisan
    - mengkopi tulisan
    - menempel tulisan
    - melihat manual Turbo Pascal
    - membatalkan operasi penyuntingan
    - keluar Turbo Pascal.

# Praktikum 2

## (Pengenalan variabel, Operasi Aritmetika, dan Input-Output)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menyebutkan jenis-jenis variabel dalam Turbo Pascal dan jangkauan nilai masing-masing jenis variabel.
4. Mahasiswa dapat membuat program Pascal untuk menampilkan nilai dengan format tertentu.
5. Mahasiswa dapat membedakan dan menggunakan perintah `write` dan `writeln` secara tepat.
6. Mahasiswa dapat menulis program Pascal sederhana yang menggunakan operasi-operasi aritmetika dan operasi bilangan bulat (modulo dan hasil bagi bulat).
7. Mahasiswa dapat menulis program yang memerlukan masukan (input).

### C. Dasar Teori

Suatu deklarasi variabel membentuk sebuah daftar pengenal yang merupakan variabel-variabel (dengan kata **var**) dan jenisnya. Jenis yang diberikan pada suatu variabel baru dapat berupa sebarang jenis yang sudah dideklarasikan sebelumnya:

- di dalam bagian deklarasi type pada blok yang sama, atau
- di dalam suatu blok tertutup, atau
- di dalam suatu **unit**.

Pengenal juga dapat berupa suatu nama jenis baru yang didefinisikan. Apabila suatu pengenal didefinisikan di dalam daftar deklarasi variabel, maka pengenal tersebut merupakan pengenal variabel untuk blok di mana deklarasi tersebut dituliskan. Variabel tersebut dapat dirujuk dari seluruh blok, kecuali pengenal tersebut dideklarasikan ulang di dalam suatu blok tertutup. Deklarasi ulang mendefinisikan suatu variabel baru dengan pengenal yang sudah ada, tanpa mempengaruhi nilai variabel aslinya.

#### Var (Kata bawaan)

Suatu deklarasi variabel (**var**) akan mengaitkan nama pengenal (variabel) yang bersangkutan dengan sebuah lokasi memori di mana nilai-nilai variabel tersebut akan disimpan dengan lebar tempat sesuai jenisnya.

#### Sintaks:

Aturan pendeklarasian variabel:

```
var
  NamaVar11, ...,NamaVar1N: jenis;
  ...
  NamaVar1M, ...,NamaVarMN: jenis;
```

### Catatan:

Kata **absolute** dapat digunakan untuk menyatakan alamat memori mutlak. Kata **absolute** dapat digunakan untuk mendeklarasikan suatu variabel absolut (yang menempati alamat memori khusus/mutlak).

Kata **var** juga dapat digunakan untuk mendeklarasikan parameter variabel (variabel yang merupakan parameter/input suatu prosedur atau fungsi).

### Contoh:

```
var
  X, Y, Z: Real;
  I, J, K: Integer;
  Angka: 0..9;
  C: Color;
  Selesai, Kesalahan: Boolean
  Operator: (Plus, Minus, Kali);
  Corak1, Corak2: set of Color;
  Hari: Date;
  Matrix: array[1..10, 1..10] of Real;
  Nama: string[15];
  Berkas1, Berkas2: Text;
  Huruf: set of 'A'..'Z';
```

### **Variabel Global dan Lokal**

*Variabel global* adalah variabel yang dideklarasikan di luar suatu prosedur atau fungsi dan menempati segmen data.

*Variabel lokal* adalah variabel yang dideklarasikan di dalam suatu prosedur atau fungsi, dan menempati segmen stack.

### Sintaks:

```
var NamaVar1: jenis absolute Segmen:Offset;
atau
```

```
var NamaVar2: jenis absolute Variabel;
```

### Catatan:

Bentuk pertama menyatakan secara langsung alamat (segmen dan offset) variabel yang bersangkutan. Bentuk kedua mendeklarasikan suatu variabel baru di atas variabel yang sudah ada (pada alamat yang sama). Daftar deklarasi pengenalan variabel harus memuat tepat sebuah nama variabel baru apabila kata **absolute** digunakan.

### Contoh:

```
type
  TabelVektor = array[0..255] of
  pointer;
var
  Vektor1: TabelVektor absolute 0:0;
  ModusCrt: Byte absolute $0040:$0049;
  Str: string;
  StrLen: Byte absolute Str;
```

Turbo Pascal mengenal lima jenis data baku: *bilangan bulat (integer)*, *bilangan nyata (real)*, *karakter dan string (char dan string)*, *nilai kebenaran logika (Boolean)*, dan *penunjuk (pointer)*.



## Data Bilangan bulat (Integer)

adalah bagian dari himpunan bilangan bulat, baik positif, nol, maupun negatif (misalnya 1, 5, -21, dan 752). Jenis data bulat dapat dikelompokkan menjadi lima macam:

Jenis Data bulat	Jangkauan Nilai	Format Penyimpanan
Shortint	-128 s/d. 127	1 byte = 8 bit (Bertanda)
Integer	-32768 s/d. 32767	2 byte = 16 bit (Bertanda)
Longint	-2147483648 s/d. 2147483647	4 byte = 32 bit (Bertanda)
Byte	0 s/d. 255	1 byte = 8 bit (tak bertanda)
Word	0 s/d. 65535	2 byte = 16 bit (tak bertanda)

Jenis data bulat termasuk jenis data terurut (type ordinal).

## Bilangan Nyata (Real)

Data **real** adalah data yang nilainya berupa bilangan nyata dan dapat dinyatakan dalam bentuk bilangan pecahan dengan desimal (misalnya, 3.14159) atau dalam bentuk eksponen (misalnya,  $2.579 \times 10^{24}$ ). Bilangan **real** juga sering disebut bilangan titik-mengambang (*floating-point number*).

Jenis data real mempunyai himpunan nilai yang berupa himpunan bagian himpunan bilangan nyata. Setiap data real dapat dinyatakan dengan notasi titik-mengambang dengan cacah digit tertentu. Dalam notasi titik-mengambang setiap nilai **N** biasanya ditulis dalam tiga bagian: **M**(antis), **B**(asis), dan **E**(ksponen) sedemikian hingga

$$\mathbf{M} \times \mathbf{B} \mathbf{E} = \mathbf{N}$$

dengan **B**=2, **M** dan **E** berupa bilangan riil bulat.

Dalam Turbo Pascal jenis data real dikelompokkan menjadi lima macam, masing-masing memiliki jangkauan nilai dan keakuratan (presisi) tersendiri, sebagaimana disajikan dalam tabel berikut ini:

Jenis Data Real	Jangkauan Nilai	Cacah Digit	Cacah Byte
real	2.9e-39 s/d. 1.7e38	11 - 12	6
single	1.5e-45 s/d. 3.4e38	7 - 8	4
double	5.0e-324 s/d. 1.7e308	15 - 16	8
extended	3.4e-4932 s/d. 1.1e4932	19 - 20	10
comp	-9.2e18 s/d. 9.2e18	19 - 20	8

**Catatan:** Jenis data **comp** merupakan jenis **integer** 64-bit dan jangkauan nilainya adalah  $(-2^{63} + 1)$  s/d.  $(2^{63} - 1)$ .

## Karakter (Char) dan String (String)

Jenis data karakter (**Char**) adalah data yang berupa sebuah huruf alfabet, simbol, dan angka 0 - 9, yakni semua karakter yang memiliki kode ASCII. Data yang berupa deretan karakter termasuk dalam jenis data **string**. Konstanta karakter (**Char**) dan **string** ditulis di antara tanda kutip tunggal,

misalnya: 'A', 'g', '?', '\*', '&', ' ', 'Turbo Pascal', 'Apakah Anda sudah makan?', dll.

## Jenis Boolean

Data jenis **Boolean** digunakan untuk menunjukkan nilai kebenaran suatu ekspresi, biasanya digunakan dalam ekspresi kondisional. Data **boolean** hanya memiliki dua kemungkinan nilai: `True` (*Benar*) atau `False` (*salah*).

## Jenis Pointer

Data jenis Pointer berupa alamat lokasi di dalam memori komputer, sebagai tempat penyimpanan informasi. Variabel jenis pointer berguna untuk menyimpan alamat memori variabel dinamik jenis tertentu.

## Operasi Aritmetika Biner

Operator	Operasi	Jenis Data (Operand)	Jenis hasil
+	Penjumlahan	integer, real	integer, real
-	Pengurangan	integer, real	integer, real
*	Perkalian	integer, real	integer, real
/	Pembagian	integer, real	real, real
<b>div</b>	Hasil bagi	integer	integer
<b>mod</b>	Sisa pembagian	integer	integer

### Catatan:

*Operator + juga digunakan sebagai operator penggabungan string dan himpunan. Operator-operator +, -, dan \* juga digunakan sebagai operator-operator union, selisih, dan irisan dalam himpunan.*

## Operator Aritmetika Uner

Operator	Operasi	Jenis Data	Jenis hasil
+	Tanda positif	integer, real	integer, real
-	Tanda negatif	integer, real	integer, real

### Catatan:

- Setiap operan (data yang dioperasikan) berjenis yang merupakan bagian dari (**subrange**) jenis ordinal diperlakukan sebagai data berjenis **ordinal**.
- Apabila kedua data dari operasi +, -, \*, **div**, **mod** merupakan salah satu jenis **integer**, maka hasil operasinya berjenis sama dengan persamaan jenis kedua operan.
- Jika salah satu atau kedua operan dari operasi +, -, atau \* berjenis **real**, hasil operasinya berjenis **real** apabila pengarah koprocesor numerik **{\$N-}** digunakan (kondisi asli) atau **extended** dalam modus **{\$N+}**.
- Apabila operasi tanda digunakan pada data jenis yang termasuk jenis integer, hasilnya berjenis integer. Jika operannya real, hasilnya berjenis **Real** atau **Extended**.
- Nilai  $x/y$  selalu berjenis **Real** atau **Extended** tanpa menghiraukan jenis operannya. Jika  $Y=0$ , maka terjadi suatu **error**.
- Nilai  $I \text{ div } J$  sama dengan hasil bagi  $I/J$ , dibulatkan ke arah nol menjadi suatu nilai berjenis integer. Jika  $J=0$ , suatu **error** akan terjadi.

- Operasi `mod` menghasilkan sisa pembagian:  

$$I \bmod J = I - (I \operatorname{div} J) * J$$
- Tanda dari hasil operasi `mod` sama dengan tanda I. Jika  $J=0$  akan terjadi suatu error.

## Perintah Pascal untuk membaca masukan (input)

### Format:

```
Read(Var1,Var2, ..., VarN);
Readln(Var1,Var2, ..., VarN);
```

Dengan perintah `Read` atau `Readln` di atas, program akan terhenti sampai pemakai memasukkan data (nilai) untuk variabel-variabel *Var1*, *Var2*, ..., dan *VarN* dan diakhiri dengan menekan tombol **ENTER**. Jenis data yang dimasukkan harus cocok dengan jenis variabel-variabel tersebut, yang ditentukan pada bagian deklarasi. Pemasukan nilai-nilai HARUS dipisahkan dengan SPASI (karakter kosong) atau dengan menekan tombol ENTER. Pada perintah `READ`, Pascal TIDAK akan berganti baris setelah pemakai menekan tombol ENTER. Pada perintah `READLN`, Pascal akan berganti baris setelah pemakai menekan tombol ENTER. Perintah

`Readln;`

akan menyebabkan program terhenti sampai pemakai menekan tombol ENTER. Hal ini bermanfaat untuk membuat jeda pada suatu proses. Perintah

```
Readln(Var1,Var2, ..., VarN);
```

ekivalen dengan perintah

```
Read(Var1,Var2, ..., VarN); Readln;
```

atau

```
Read(Var1); Read(Var2); ...;Read(VarN);readln;
```

### Contoh:

```
read(a);read(b);read(c);
readln>Nama,Nilai1,Nilai2,Nilai3);
read(Alamat,TglLahir);readln;
```

## Perintah Pascal untuk mencetak tampilan string atau nilai variabel.

Tata tulis perintah:

```
Write(Nilai1:lebar,Nilai2:lebar, ...);
```

```
Writeln(Nilai1:lebar,Nilai2:lebar, ...);
```

dengan *Nilai1*, *Nilai2*, dst. dapat berupa konstanta (untuk konstanta **Char** atau **String** harus diapit dengan tanda petik tunggal), atau nilai suatu variabel. Tambahan (sifatnya opsional) *:lebar* berguna untuk menentukan lebar tempat untuk mencetak *Nilai1*, *Nilai2*, dst. dan menyebabkan *Nilai1*, *Nilai2*, dst. dicetak rata kanan dengan tempat selebar *lebar*. Apabila *Nilai1* merupakan bilangan jenis **real**, maka format

```
Nilai1:lebar1:lebar2
```

akan mencetak *Nilai1* dengan menggunakan *lebar1* tempat untuk bagian bulat dan *lebar2* tempat untuk bagian pecahan desimal.

Perintah `Write(..)` tidak menyebabkan pergantian baris setelah program menampilkan nilai. (Kursor akan tetap berada di belakang karakter terakhir yang dicetak.) Perintah `Writeln(..)` akan menyebabkan pergantian baris setelah mencetak nilai. (Kursor akan berpindah ke kolom pertama baris di bawahnya.)

## Perintah

`writeln;`

akan mencetak sebuah baris kosong.

## Contoh:

Perintah	Tampilan
<code>write('Selamat Datang!');</code>	Selamat Datang!
<code>nama:='Agus'; Nilai:=6.754; writeln(nama:10,nilai:3:1); write('Ada 6 spasi kosng di depan Agus');</code>	Agus 6.8 Ada 6 spasi kosong di depan Agus
<code>write(1);writeln(2);writeln;writeln(3);</code>	12 3

Perintah berikut ini berguna untuk mengatur tampilan (output) di layar.

ScrollTo	<code>ScrollTo(X, Y: Integer);</code>	Menggulung layar ke posisi (X,Y). Posisi (0,0) adalah pojok kiri atas layar.
gotoXY	<code>gotoXY(X,Y:Integer);</code>	Memindahkan kursor ke posisi (X,Y)

## D. Kegiatan Praktek

1. Tulis program-program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----  
{1} program tutor2;  
{2} uses wincrt;  
{3} var x,y,z:integer;  
{4} begin  
{5}     x:=3; writeln('x = ',x);  
{6}     y:=x*2; writeln('y = ',y);  
{7}     x:=y-5;writeln('x = ',x);  
{8}     z:=x+y; writeln('z = ',z)  
{9}     writeln(z,'-',x,'=',z-x);  
{10}    writeln(z,'+',x,'=',z+x);  
{11}    writeln(z,'x',x,'=',z*x);  
{12} end.
```

```
-----  
{1} program operasil;  
{2} uses wincrt;  
{3} var bagi, hasilbagi, sisa: integer;  
{4}   begin  
{5}     bagi:=15/4;  
{6}     hasilbagi:=15 div 4;  
{7}     sisa:= 15 mod 4;  
{8}     writeln('Hasil pembagian 15 oleh 4 adalah ',bagi);  
{9}     writeln('Hasilbagi bulat 15 oleh 4 adalah ',  
             hasilbagi);  
{10}    writeln('Sisa pembagian 15 oleh 4 adalah ', sisa);
```

```
{11} end.
```

2. Tulis program-program di bawah ini (persisi apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----  
program teks1;  
uses wincrt;  
var teks: string;  
begin  
    writeln('Ketik suatu tulisan (teks) dan akhiri dengan  
           tombol ENTER:');  
    readln(teks);  
    write('Anda tadi menuliskan: ',#10#13,teks);  
end.
```

```
-----  
program operasi2;  
uses wincrt;  
var bil1, bil2:integer;  
begin  
    write('Masukkan sebuah bilangan: ');  
    read(bil1);  
    write('Masukkan bilangan kedua: ');  
    readln(bil2);  
    writeln(bil1,' + ',bil2,' = ',bil1+bil2);  
    writeln(bil1,' - ',bil2,' = ',bil1-bil2);  
    writeln(bil1,' x ',bil2,' = ',bil1*bil2);  
    writeln(bil1,' : ',bil2,' = ',bil1/bil2:6:2);  
    writeln(bil1,' DIV ',bil2,' = ',bil1 div bil2);  
    writeln(bil1,' MOD ',bil2,' = ',bil1 mod bil2);  
end.
```

3. Tulis program di bawah ini (persisi apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----  
Program SPL2v;  
uses wincrt;  
var a,b,c,d,e,f,det,x,y: real;  
begin  
    writeln('Menyelesaikan SPL ax + by = c');  
    writeln('           dx + ey = f') ;  
    writeln;  
    write('Masukkan nilai a, b, c, d, e, f ');  
    writeln('(pisahakan dgn SPASI/ENTER):');  
    readln(a,b,c,d,e,f);  
    det:= (a*e-b*d) ;  
    if det <> 0 then  
        begin  
            x:= (c*e-b*f)/det; y:= (a*f-c*d)/det;  
            write('Penyelesaikan SPL di atas adalah (');  
            writeln(x:5:2,',',y:5:2,')');  
        end  
    else if (a/d <> c/f) then writeln('SPL di atas tidak konsisten!')  
    else  
        begin
```

```

        writeln('SPL di atas mempunyai tak berhingga penyelesaian');
        write('Bentuk umum penyelesaiannya adalah ');
        writeln('x = ',-b/a:5:2,'y +',c/a:5:2);
    end;
writeln('Tekan ENTER jika sudah paham ...');readln;
clrscr;
writeln('Terima kasih ... sampai ke Program berikutnya.') ;
writeln('Tekan ENTER untuk keluar program ...');readln;
donewincrt;
end.
-----

```

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

A. Judul Praktikum : **Praktikum Pemrograman Komputer**

B. Identitas Praktikan :

Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_

Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_

C. Hasil Praktikum :

1. a). Apa yang terjadi jika program **tutor2** dikompilasi atau dijalankan? b). Pada baris manakah terjadi kesalahan? c). Mengapa? Betulkan! d). Tuliskan tampilan programnya. e). Mengapa tampilan pada baris pertama berbeda dengan tampilan pada baris ketiga?
  - f). Ubah semua perintah `writeln` menjadi `write`. Tuliskan kembali tampilan programnya. g). Apa perbedaan perintah `write` dan `writeln`?
2. a). Mengapa program `operasil` tidak dapat jalan? b). Bilangan apakah hasil pembagian  $15/4$ , bulat atau pecahan? c). Bilangan pecahan harus disimpan ke dalam variabel jenis apa, **integer** atau **real**? Ubah definisi variabel bagi menjadi `real`:
 

```
var bagi: real;
```

  - d). Jalankan kembali dan perhatikan serta catat hasilnya.
  - e). Ubah baris ke-8 menjadi
 

```
writeln('Hasil pembagian 15 oleh 4 adalah ',bagi:3:4);
```

 Tuliskan tampilan pada baris pertama jika program dijalankan kembali.
  - f). Apa artinya tambahan `:3:4` tersebut?
3. a) Jalankan program `teks1`. Ketik beberapa kalimat. b) Ulangi menjalankan program tersebut. Cobalah menulis beberapa kalimat yang panjang. c) Berapa karakterkah (termasuk spasi) maksimum Anda dapat menuliskan teks masukan?

4. a). Jalankan program operasi2 berkali-kali dan masukkan dua buah bilangan, kemudian buat dan lengkapi tabel seperti di bawah ini.

Bil1	Bil2	Tampilan					
		Baris ke- 3	Baris ke- 3	Baris ke- 3	Baris ke- 3	Baris ke- 3	Baris ke- 3
179	45						
0	7500						
32100	12345						
150	3						

- b) Jawaban manakah yang benar dan manakah yang salah? Mengapa terjadi salah jawaban?
- c). Ubah definisi variabel bil1 dan bil2 menjadi shortint. Jalankan kembali programnya berkali-kali dengan data pada tabel di atas. Jawaban manakah yang salah? Mengapa demikian?
- d). Ubah definisi variabel bil1 dan bil2 menjadi longint. Jalankan kembali programnya berkali-kali dengan data pada tabel di atas. Jawaban manakah yang salah? Mengapa demikian?
- e). Ubah definisi variabel bil1 dan bil2 menjadi real. Jalankan kembali programnya berkali-kali dengan data pada tabel di atas. Jawaban manakah yang salah? Mengapa demikian?
5. a). Tulis sebuah program untuk menghitung keliling (K) dan luas lingkaran (L) dengan rumus  $K=2\pi R$  dan  $L=\pi R^2$ , dengan  $\pi=3.14$ . Simpan program Anda dengan nama lingkarn.pas. b). Tentukan jenis variabel apakah yang tepat untuk nilai-nilai K, L, dan R?Lengkapi tabel berikut dengan memberi tanda centang pada isian yang tepat:

Variabel \ Jenis	Shortint	Integer	longint	Real
R				
K				
L				

- c). Jalankan program Anda beberapa kali dan lengkapi tabel berikut.

Masukan Nilai R	1	2	2.5	3	5	10
--------------------	---	---	-----	---	---	----

Keliling (K)						
Luas (L)						

Ketik program di bawah ini dan simpan dengan nama segi3lk.pas. Bandingkan hasil program Anda dengan hasil program di bawah ini. Perhatikan perintah-perintah Pascal baru dan catat kegunaannya!

```

-----
Program segi3lk;
uses wincrt;
var r,a,b,c,l,k: real;
n:integer;
begin
  writeln('Menghitung luas segitiga siku-siku');
  writeln('Masukkan panjang sisi-sisi tegak:');
  write('a = ');readln(a); write('b = ');readln(b);
  c:=sqrt(a*a+b*b); {sisi miring}
  l:=a*b/2;      {luas segitiga}
  writeln('OK! Luas segitiga tersebut adalah', l:5:3);
  writeln('sedangkan panjang sisi miringnya = ',c:5:3);
  writeln;
  writeln('Tekan ENTER untuk melanjutkan...');readln;
  clrscr; {bersihkan layar ...}
  writeln('Menghitung keliling dan luas lingkaran jika
diketahui jari-jarinya');
  write('Panjang jari-jari lingkaran, R = '); readln(r);
  k:=2*pi*r; {keliling lingkaran}
  l:=pi*r*r; {luas lingkaran}
  writeln('Keliling lingkaran: K = ', k:5:3);
  writeln('Luas lingkaran: L = ', l:5:3);
  writeln;
  writeln('Tekan ENTER jika sudah paham ...');readln;
  clrscr;
  writeln('Selesai. Mau keluar? Tekan tombol ENTER!') ;
  readln; donewincrt;
end.
-----

```



# Praktikum 3

## (Label, Goto, Operator Logika, IF, dan CASE)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menulis program Pascal yang menggunakan perintah IF ... THEN ... dan IF ... THEN ... ELSE ....
4. Mahasiswa dapat membuat program Pascal yang menggunakan perintah LABEL dan GOTO.
5. Mahasiswa dapat menulis program Pascal yang menggunakan struktur CASE .. OF ... dan CASE ... OF ... ELSE .... END.
6. Mahasiswa dapat menulis program Pascal untuk menghitung akar persamaan kuadrat.
7. Mahasiswa dapat menulis program untuk membandingkan dua buah bilangan bulat.
8. Mahasiswa dapat menulis program Pascal untuk melakukan operasi dua buah bilangan.

### C. Dasar Teori

Operator logika (**boolean**) dalam Pascal meliputi:

Operator	Operasi	Jenis Data	Jenis Hasil
<b>not</b>	negasi	Boolean	Boolean
<b>and</b>	DAN	Boolean	Boolean
<b>or</b>	ATAU	Boolean	Boolean
<b>xor</b>	XOR	Boolean	Boolean

Operator **not** merupakan operator uner.

Operator relasi dalam Pascal adalah sebagai berikut.

Operator	Operasi	Jenis Data	jenis hasil
=	sama dengan	semua jenis baku, pointer, set, string, atau packed string	Boolean
<>	tidak sama	semua jenis baku, pointer, set, string, atau packed string	Boolean
<	kurang dari	semua jenis baku, string, packed string, atau PChar	Boolean
>	lebih dari	semua jenis baku, string, packed string, atau PChar	Boolean
<=	kurang dari atau sama dengan	semua jenis baku, string, packed string, atau PChar	Boolean
>=	lebih dari atau sama dengan	semua jenis baku, string, packed string, atau PChar	Boolean

### Peluncutan program dengan GOTO

Perintah

```
goto NamaLabel;
```

digunakan untuk membuat agar jalannya program meloncat ke statemen yang diawali dengan kata NamaLabel: (**ada titik duanya!**). NamaLabel harus didefinisikan pada bagian deklarasi dengan kata label.

### Contoh:

```
label baca;
begin
  baca:
  read(a);
  if a = 0 then goto baca;
  ...
end.
```

### Pengendalian program dengan if ... then ... else

Kata **if**, **then**, dan **else** digunakan untuk mengendalikan jalannya suatu program berdasarkan pada suatu kondisi atau syarat. Perintah ini berguna untuk memilih statemen mana yang akan dikerjakan oleh komputer berdasarkan nilai kebenaran syarat yang diberikan. Format `if ... then ... else` adalah:

```
if kondisi then statemen;
atau
if kondisi then statemen1 else statemen2;
```

Apabila *kondisi* bernilai benar (True) maka *statemen* (atau *statemen1*) akan dikerjakan. Jika *kondisi* bernilai salah (False) maka *statemen* (atau *statemen1*) TIDAK dikerjakan. Apabila ada **else** maka yang dikerjakan adalah statemen setelah **else**, yakni *statemen2*.

### Catatan:

Apabila kata **else** digunakan, maka statemen antara **then** dan **else** TIDAK diakhiri dengan titik koma, tetapi statemen setelah **else** harus diakhiri dengan titik koma (;). Apabila tidak ada kata **else**, maka statemen setelah **then** HARUS diakhiri dengan titik koma (;).

### Contoh:

```
(* Statemen "if - then - else" *)
if (I < Min) or (I > Max) then I := 0;

if ParamCount <> 2 then
begin
  WriteLn('Salah tulis perintah');
  Halt(1); {Program terhenti ...}
end
else
begin
  ReadFile(ParamStr(1));
  WriteFile(ParamStr(2));
end;
```

### Pengendalian program dengan case..of..else..end

Statemen **case** terdiri atas sebuah *ekspresi* (pemilih) dan sebuah daftar *statemen*, masing-masing diawali dengan sebuah nilai (*kasus ekspresi*) dan titik dua (:).

### Aturan pemakaian:

```
case ekspresi of
  nilai1: statement1;
```

```

    nilai2: statement2;
    ...
end

```

atau

```

case ekspresi of
    nilai1: statement1;
    ...
    nilaiN: statementN;
else
    statementLain;
end

```

Pada pola pertama, *statement1* akan dikerjakan apabila nilai *ekspresi*=nilai1. Jika tidak, *statement2* akan dikerjakan apabila nilai *ekspresi*=nilai2, dst. Pada pola kedua, sama dengan pola pertama, kecuali bahwa *statementLain* akan dikerjakan apabila nilai *ekspresi* tidak sama dengan nilai1, nilai2, ..., maupun nilaiN.

Nilai *ekspresi* harus suatu ekspresi yang nilainya berjenis ordinal (dapat diurutkan). Nilai1, Nilai2, dst. dapat berupa sebuah nilai jenis ordinal (sesuai dengan jenis *ekspresi*) atau beberapa nilai yang dipisahkan dengan tanda koma.

### Contoh:

```

case karakter of
    'A'..'Z', 'a'..'z': WriteLn('Huruf');
    '0'..'9':           WriteLn('Angka satuan');
    '+', '-', '*', '/': WriteLn('Operator');
else
    WriteLn('Karakter khusus!');
end;

```

## D. Kegiatan Praktek

1. Tulis program-program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program tutor3;
uses wincrt;
var x,y:integer;
pilih:char;
label baca;
begin
baca:
clrscr;
writeln('Program Membaca Dua Bilangan Bulat');
write('Masukkan bilangan pertama: ');readln(x);
write('Masukkan bilangan kedua: ');readln(y);
if x > y then
writeln('Anda tentu tahu bahwa ', x, ' > ', y)
else
writeln('Anda tentu tahu bahwa ', x, ' <= ', y);
writeln;
write('Mau melanjutkan? <Y/T>: ');readln(pilih);
if (pilih = 'Y') or (pilih = 'y') then goto baca;
writeln('Program sudah selesai ... terima kasih!');

```

end.

```
-----  
Program Operasi3;  
uses wincrt;  
var x,y:integer;  
operasi,pilih:char;  
label baca;  
begin  
  baca:  
  clrscr;  
  writeln('Program Operasi Dua Bilangan Bulat');  
  write('Masukkan bilangan pertama: ');readln(x);  
  write('Masukkan bilangan kedua: ');readln(y);  
  write('Pilih Operasi : ');readln(operasi);  
  if operasi='+' thenwriteln(x,' + ', y, ' = ', x+y)  
  else  
  if operasi='- ' thenwriteln(x,' - ', y, ' = ', x-y)  
    else  
      if (operasi='*') or (operasi='x') or (operasi='X')  
        thenwriteln(x,' * ', y, ' = ', x*y)  
        else  
          if ((operasi='/') or (operasi=':')) and (y<>0) then  
writeln(x,' : ', y, ' = ', x/y:4:3)  
          else  
writeln('Error! Anda tidak tahu operasi aritmetika?');  
writeln;  
write('Mau melanjutkan? <Y/T>: ');readln(pilih);  
if (pilih = 'Y') or (pilih = 'y') then goto baca;  
writeln('Program sudah selesai ... terima kasih!');  
end.  
-----
```

```
Program AkarPK;  
uses WinCrt;  
var  
  a,b,c:real;  
  D,x1,x2:real;  
  R,I:real;  
  pilih:Char;  
label baca;  
begin  
  baca:  
  clrscr;  
  writeln('Program menghitung akar persamaan  $ax^2+bx+c=0$ ');  
  writeln;  
  write('Masukkan nilai-nilai a, b, dan c :');  
  readln(a,b,c);  
  if a = 0 then  
  begin  
    writeln('Nilai a harus tidak nol!');  
    goto baca; {Ulangi lagi sampai a<>0!}  
  end;  
  (* Tulis persamaan ... *)  
  writeln('Anda mempunyai persamaan kuadrat: ');
```

```

if a=1 then write('x^2') else write(a:3:2,'x^2');
if b>0 then
  begin
    if b=1 then write(' + x')
    else write(' + ',b:3:2,'x')
  end
  else if b<0 then write(b:3:2,'x');
if c>0 then write(' + ',c:3:2)
  else if c<0 then write(c:3:2);
writeln(' = 0. ');writeln;
D:=b*b-4*a*c; {dapat juga ditulis D:=sqr(b)-4*a*c}
if D=0 then
  begin
    x1:=-b/(2*a);
    writeln('Oleh karena nilai D = ',D:3:2,', ');
    writeln('maka akar-akarnya adalah x1 = x2 = ',x1:3:2);
  end
else if D>0 then
  begin
    x1:=(-b+sqr(D))/(2*a);x2:=(-b-sqr(D))/(2*a);
    writeln('Oleh karena nilai D = ',D:3:2,', ');
    writeln('maka akar-akarnya adalah:');
    writeln('x1 = ',x1:3:2,' dan x2 = ',x2:3:2);
  end
else if D<0 then
  begin
    R:=-b/(2*a);I:=(sqr(-D))/(2*a);
    writeln('Oleh karena nilai D = ',D:3:2,', ');
    writeln('maka akar-akarnya adalah:');
    write('x1 = ');
    if R<>0 then write(R:3:2);
    if I>0 then write('+',I:3:2,'i')
  else write(I:3:2,'i');
    write(' dan x2 = ');
    if R<>0 then write(R:3:2);
    if i>0 then writeln('-',I:3:2,'i')
  else writeln(I:3:2,'i');
  end;
writeln;
write('Mau menghitung yang lain? <Y/T>: ');
readln(pilih);
if (pilih='Y') or (pilih='y') then goto baca
  else writeln('Selesai!');
donewincrt;
end.
-----
Program Operasi4;
uses wincrt;
var x,y:integer;
operasi,pilih:char;
label baca;
begin
  baca:

```

```

clrscr;
writeln('Program Operasi Dua Bilangan Bulat');
write('Masukkan bilangan pertama: ');readln(x);
write('Masukkan bilangan kedua: ');readln(y);
write('Pilih Operasi : ');readln(operasi);
case operasi of
  '+' : writeln(x,' + ', y, ' = ', x+y);
  '-' : writeln(x,' - ', y, ' = ', x-y);
  '*', 'x', 'X': writeln(x,' * ', y, ' = ', x*y);
  '/', ':': if (y<>0) then
writeln(x,' : ', y, ' = ', x/y:4:3);
else
writeln('Error! Anda tidak tahu operasi aritmetika?');
end;
writeln;
write('Mau melanjutkan? <Y/T>: ');readln(pilih);
if (pilih = 'Y') or (pilih = 'y') then goto baca;
writeln('Program sudah selesai ... terima kasih!');
end.
-----

```

### E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

- A. Judul Praktikum : **Praktikum Pemrograman Komputer**
- B. Identitas Praktikan :
  - Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_
  - Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_
- C. Hasil Praktikum :

1. a) Jalankan program `tutor3.pas` beberapa kali dan lengkapi tabel di bawah ini.

No	Bil. I	Bil. II	Tampilan relasi kedua bilangan
1	32400	5657	
2	-6580	0	
3	-33000	5000	
4	5.3	5	
5	7500	100000	
6	9	4+5	
7	-50000	-32000	
8	65321	12345	

- b). Tampilan manakah yang benar?
- c). Tampilan manakah yang benar tetapi bilangannya berubah?
- d). Mengapa terjadi perubahan nilai bilangan yang ditampilkan? Bagaimanakah cara agar tidak terjadi hal demikian?
- e). Nomor-nomor manakah yang menampilkan hasil yang salah dan berubah bilangannya? Bagaimanakah untuk menghindari terjadinya hal demikian?

- f). Manakah yang menyebabkan terjadinya pesan **error** ? Mengapa demikian? Bagaimanakah caranya untuk menghindari hal demikian?
- g). Dapatkah kesalahan yang terjadi pada isian nomor 6 dihindari dengan cara mengubah programnya?
- h). Apakah kegunaan perintah label dan bagaimanakah cara menggunakannya?

2. a). Jalankan programOperasi3.pas beberapa kali dan lengkapi tabel di bawah ini.

No	Bil. I	Bil. II	Operasi	Tampilan yang dihasilkan
1	23	45	+	
2	23	45	-	
3	35	5	/	
4	35	5	:	
5	35	5	X	
6	35	10	*	
7	500	500	x	
8	15	0	/	
9	0	15	/	
10	47	40	<	
11	32000	32000	+	
12	32000	32000	-	
13	-3210	65000	+	
14	350	100	*	
15	50.0	10	/	

- b). Nomor-nomor manakah yang menghasilkan tampilan (hasil perhitungan) yang salah?
- c). Nomor-nomor manakah yang menghasilkan pesan **error** (tidak ada hasil perhitungan)?
- d). Bagaimanakah cara menghindari kesalahan seperti nomor 7 dan 14?
- e). Bagaimanakah cara menghindari kesalahan seperti nomor 8?
- f). Bagaimanakah cara menghindari kesalahan seperti nomor 9?
- g). Mengapa nomor 11 hasilnya salah padahal nomor 12 hasilnya benar?
- h). Bagaimanakah cara menghindari kesalahan seperti nomor 11?
- i). Bagaimanakah cara menghindari kesalahan seperti nomor 13?
- j). Bagaimanakah cara menghindari kesalahan seperti nomor 15?

3. a). Jalankan programAkarPK.pas beberapa kali dan lengkapi tabel di bawah ini.

No	a	b	c	D	x1	x2	Jelaskan hasilnya
1	1	1	1				
2	0	1	2				
3	-1	3	2				
4	2	3	0				
5	1	0	0				
6	-1	5	10				
7	0	1	0				
8	2	3	4				
9	1	2	-10				
10	1	-5	6				
11	3/2	1	4.0				
12	1.5	1	4.0				
13	2.5	5	1.0				
14	-0.5	2	4				
15	-3	5	10				

- b). Nomor-nomor manakah yang menghasilkan akar kembar? Mengapa?
- c). Nomor-nomor manakah yang menghasilkan akar kompleks? Mengapa?
- d). Nomor-nomor manakah yang menghasilkan dua akar berlainan? Mengapa?

- 4. Bagaimanakah cara memberi masukan (input) untuk merespon perintah `read(a, b, c)`?
- 5. Apakah kegunaan perintah `clrscr`?
- 6. a). Jalankan program `Operasi4.pas` beberapa kali dan lengkapi tabel di bawah ini.

No	Bil. I	Bil. II	Operasi	Tampilan yang dihasilkan
1	23	45	+	
2	23	45	-	
3	35	5	/	
4	35	5	:	
5	35	5	X	
6	35	10	*	
7	500	500	x	
8	15	0	/	
9	0	15	/	
10	47	40	<	
11	32000	32000	+	
12	32000	32000	-	
13	-3210	65000	+	
14	350	100	*	
15	50.0	10	/	

- b). Nomor-nomor manakah yang menghasilkan tampilan (hasil perhitungan) yang salah?
- c). Nomor-nomor manakah yang menghasilkan pesan **error** (tidak ada hasil perhitungan)?
- d). Bagaimanakah cara menghindari kesalahan seperti nomor 7 dan 14?
- e). Mengapainomor 8 tidak menampilkan kesalahan? Bagaimanakah untuk memberi tahu terjadinya pembagian dengan nol?
- f). Bagaimanakah cara menghindari kesalahan seperti nomor 9?
- g). Mengapa nomor 11 hasilnya salah padahal nomor 12 hasilnya benar?
- h). Bagaimanakah cara menghindari kesalahan seperti nomor 11?
- i). Bagaimanakah cara menghindari kesalahan seperti nomor 13?
- j). Bagaimanakah cara menghindari kesalahan seperti nomor 15?
- 7. Apakah kesamaan dan perbedaan fungsi dan pemakaian perintah `IF ... THEN ... ELSE` dan `CASE ... OF ... ELSE ...`?



# Praktikum 4

## (Pengulangan Proses: *Looping* dengan FOR ... TO ... DO, WHILE ... DO, ... REPEAT ... UNTIL ...)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menyebutkan jenis-jenis variabel dalam Turbo Pascal dan jangkauan nilai masing-masing jenis variabel.
4. Mahasiswa dapat membuat program Pascal untuk melakukan pengulangan proses.
5. Mahasiswa dapat membedakan dan menggunakan struktur pengulangan (*loop*) dengan *for .. to .. do*, *while .. do*, dan *repeat .. until*.
6. Mahasiswa dapat menulis program Pascal sederhana yang menggunakan struktur pengulangan (*loop*) dengan *for .. to .. do*, *while .. do*, dan *repeat .. until*.
7. Mahasiswa dapat menulis program yang memerlukan masukan (input) secara berulang.

### C. Dasar Teori

Ada tiga struktur di dalam Pascal untuk melakukan pengulangan proses (*looping*), yakni: (1) *for..to* atau *for ..downto..do*, (2) *while..do*, dan (3) *repeat..until*.

#### Loop dengan *for ... to*, *for ... downto*

##### Sintaks:

```
for variabel:=nilai1 to nilai2 do statment;  
atau  
for variabel:=nilai2 downto nilai1 do statment;
```

Statemen *for* (dengan syarat  $nilai2 > nilai1$ ) di atas mengakibatkan *statement* setelah kata *do* dikerjakan sebanyak ( $nilai2 - nilai1 + 1$ ) kali, dari nilai *variabel=nilai1* sampai *variabel=nilai2* (atau sebaliknya). Dalam hal ini, *variabel* merupakan variabel indeks (penghitung). *variabel*, *nilai1*, dan *nilai2* harus termasuk jenis ordinal (yang nilainya dapat dihitung secara berurutan, misalnya *Char* dan *Integer*).

##### **Contoh:**

```
(* for ... to, for ... downto *)  
for I := 1 to 10 do WriteLn(I); {cetak 1, 2, ..., 10 ke bawah}  
for I := 1 to 10 do  
  for J := 1 to 10 do  
  begin  
    X := 0;  
    for K := 1 to 10 do  
      X := X + M1[I, K] * M2[K, J];  
    M[I, J] := X; {simpan nilai-nilai X ke dalam matriks M berukuran 10 x 10}  
  end;
```

## Loop dengan while .. do

### Sintaks:

```
while ekspresi do statemen;
```

Struktur tersebut akan mengakibatkan *statemen* dikerjakan selama *ekspresi* bernilai `True`. Nilai kebenaran akan *ekspresi* dicek sebelum *statemen* dikerjakan. Jika *ekspresi* bernilai `False` di awal (sebelum `while`), maka *statemen* tidak akan pernah dikerjakan.

### Contoh:

```
I:=1;
while I < 10 do
  begin
    WriteLn(I); {cetak 1, 2, ..., 9 ke bawah}
    I:=I+1;
  end;
```

## Loop dengan repeat .. until ..

### Sintaks:

```
repeat
  statement1;
  statement2;
  ...
  statement
until ekspresi;
```

Struktur tersebut akan menyebabkan *statemen-statemen* di antara kata **repeat** dan **until** dikerjakan selama *ekspresi* Boolean setelah kata `until` bernilai `True`. *Statemen-statemen* antara kata **repeat** dan **until** akan dikerjakan sekali, kemudian *ekspresi* setelah `until` dicek, jika bernilai `False`, maka proses pengulangan terhenti.

### Contoh:

```
I:=1;
repeat
  WriteLn(I); {cetak 1, 2, ..., 9 ke bawah}
  I:=I+1;
until I=10;
```

## D. Kegiatan Praktek

1. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----
Program Tblkali;
uses wincrt;
var
  dimensi, i, j: integer;
begin
  writeln('Mencetak tabel perkalian');
  write('Masukkan ukuran tabel (1 .. 15): ');
  readln(dimensi); writeln;
  if (dimensi >= 1) and (dimensi <= 15) then
    begin
```

```

write('*|':3); {judul tabel & pembatas judul}
for i := 1 to dimensi do write(i:4);
writeln;
for i := 1 to (dimensi+1)*4 do write('=');
writeln;
for i := 1 to dimensi do {cetak tabel }
begin
    if i < 10 then write(' ');
    write(i,'|');
    for j := 1 to dimensi do write(i*j :4);
    writeln;
end;
end
else writeln('Anda salah memasukkan ukuran!');
end.

```

2. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

-----  
Program loopfor;

```

uses wincrt;
var
    angka1,angka2,angka3 : integer;
    huruf1,huruf2         : char;
begin
    for angka1:=1 to 10 do write(angka1:2);
    writeln;
    for angka2:=10 downto 1 do write(angka2:2);
    writeln;
    for huruf1:='A' to 'P' do write(huruf1:2);
    writeln;
    for huruf2:='z' downto 's' do write(huruf2:2);
    writeln;
    for angka3:=175 to 200 do write(chr(angka3):2);
end.

```

-----

3. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

-----  
Program Pgktblt;

```

uses wincrt;
var basis,pangkat,hasil,i:integer;
begin
    writeln('Program Perpangkatan Bilangan Bulat');
    write('Tulis basis: ');readln(basis);
    write('Tulis pangkat: ');readln(pangkat);
    hasil:=1;
    i:=1;
    while i<pangkat do
        begin

```

```

        inc(i);
        hasil:=hasil*basis;
    end;
    writeln(basis,'^',pangkat,' = ', hasil:6);
    writeln(basis,'^',pangkat,' = ',
        exp(pangkat*ln(basis)):5:3,'(dengan rumus a^x =
        e^[x ln (a)])');
end.
-----

```

4. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
program fktorial;
uses wincrt;
var
    i,n: byte;
    faktorial: integer;
begin
    writeln('Menghitung nilai faktorial');
    write('Masukkan bilangan asli n = ');readln(n);
    faktorial:=1;
    for i:= 1 to n do faktorial:=faktorial*i;
    writeln(n,'! = ',faktorial);
end.
-----

```

5. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
program dec2bin;
uses wincrt;
var
    n : integer;
    sisa : integer;
    biner,bit:string;
begin
    writeln('Konversi sistem desimal ke biner. ');
    write('Masukkan bilangan asli n = ');readln(n);
    biner:='';
    sisa:=0;
    while n>0 do
        begin
            sisa:=n mod 2;
            n:= n div 2;
            str(sisa,bit);
            biner:=bit+biner;
        end;
    writeln(n, ' dalam sistem biner = ',biner);
end.
-----

```

6. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----  
program FPBnKPK;  
uses wincrt;  
var  
    a,b,r,m,n : integer;  
begin  
    writeln('Mencari FPB dan KPK');  
    write('Masukkan bilangan asli pertama a = '); readln(a);  
    write('Masukkan bilangan asli kedua b = '); readln(b);  
    m:=a;n:=b;  
    repeat  
        r:= m mod n;  
        if (r<>0) then  
            begin  
                m:=n;  
                n:=r;  
            end;  
    until (r=0);  
    writeln('FPB(' , a , ',' , b ,') = ',n);  
    writeln('KPK(' , a , ',' , b ,') = ',a*b div n);  
end.  
-----
```

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

A. Judul Praktikum : **Praktikum Pemrograman Komputer**

B. Identitas Praktikan :

Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_

Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_

C. Hasil Praktikum :

1. **Compile** program-program yang Anda tulis tersebut. Cek apakah terdapat kesalahan. Jika masih salah, betulkan. Jika sudah benar, cek dengan masukan yang sesuai, apakah outputnya benar.
2. Setelah program Anda benar error, jalankan program-program Anda di atas beberapa kali dengan data yang berbeda-beda, kemudian tuliskan hasilnya pada bagian ini.
3. Tuliskan perintah-perintah Pascal baru yang baru saja Anda ketahui dan jelaskan kegunaannya.

## Praktikum 5 (Penggunaan Fungsi-fungsi Matematika)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menyebutkan jenis-jenis variabel dalam Turbo Pascal dan jangkauan nilai masing-masing jenis variabel.
4. Mahasiswa dapat membuat program Pascal untuk menghitung nilai-nilai beberapa fungsi matematika yang tersedia pada Turbo Pascal.
5. Mahasiswa dapat membedakan dan menggunakan fungsi-fungsi matematika dalam Turbo Pascal untuk menghitung nilai-nilai fungsi matematika umum.
6. Mahasiswa dapat menulis program Pascal sederhana yang menggunakan operasi-operasi aritmetika dan fungsi-fungsi matematika.
7. Mahasiswa dapat menulis program yang memerlukan masukan (input) untuk menghitung nilai fungsi.

### C. Dasar Teori

Berikut adalah daftar fungsi/prosedur matematika yang sudah disediakan oleh Turbo Pascal.

Nama Fungsi	Pemakaian	Hasil
Abs	Abs(X): (saja jenis dengan jenis X);	X
ArcTan	ArcTan(X: Real): Real; <b>Catatan:</b> Turbo Pascal tidak menyediakan fungsi Tan, karena fungsi ini dapat dihitung dengan pembagian Sin(x) / Cos(x)	Tan <sup>-1</sup> (X)
Cos	Cos(X: Real): Real;	Cos(x)
Dec	Dec(var X[ ; N: Longint]);	Mengurangi nilai X (dan N) dengan 1.
Exp	Exp(X: Real): Real;	e <sup>x</sup>
Frac	Frac(X: Real): Real;	Bagian pecahan pada X
Inc	Inc(var X [ ; N: Longint ] );	Menambah nilai X (dan N) dengan 1
Int	Int(X: Real): Real;	Bagian bulat pada X
Ln	Ln(X: Real): Real;	Ln(x)
Odd	Odd(X: Longint): Boolean;	Menguji apakah X ganjil
Ord	Ord(X): Longint;	Nilai ordinal (urutan/posisi X)
Pi	Pi: Real;	Nilai π, yakni 3.1415926535897932385.
Pred	Pred(X): < sejenis dengan jenis X >;	Nilai pendahulu X (dalam urutan)

Random	Random(X: word): < sejenis dengan parameter>; <b>Catatan:</b> Bilangan acak harus dibangkitkan dengan perintah <b>Randomize</b> , atau memberi nilai <b>RandSeed</b> .	Menghasilkan bilangan acak bulat antara nol dan X (nol dan X mungkin terjadi)
Round	Round(X: Real): Longint;	Pembulatan X
Sin	Sin(X: Real): Real;	Sin(x)
Sqr	Sqr(X):< sejenis dengan parameter>;	$X^2$
Sqrt	Sqrt(X: Real): Real;	$\sqrt{X}$
Succ	Succ(X): < sejenis dengan parameter>;	Nilai setelah X (dalam urutan)
Trunc	Trunc(X: Real): Longint;	Mengambil bagian bulat X sebagai jenis <b>longint</b>

## D. Kegiatan Praktek

1. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----
program Tebakbil;
uses wincrt;
var
    pilih: char;
    bil_acak,tebakan,kesempatan: integer;
label mulai;
begin
    mulai:
    clrscr;
    randomize; {panggil pembangkit bilangan acak}
    bil_acak := random(100) + 1; {hasilkan sebuah
                                bilangan acak yang harus ditebak}
    kesempatan := 6; {banyaknya kesempatan menebak}
    tebakan := 1000; {beri nilai awal tebakan}
    writeln('Saya mempunyai bilangan antara 1 dan 100. ');
    writeln('Cobalah Anda tebak!');
    while (kesempatan > 0) and (tebakan <> bil_acak) do
        {selagi masih punya kesempatan menebak}
    begin
        readln(tebakan);
        if tebakan = bil_acak then
            writeln('Tepat sekali, Anda memang hebat!')
        else
            begin
                dec(kesempatan);
                if kesempatan = 0 then
                    begin
                        writeln('Maaf ya, Anda kehabisan kesempatan menebak. ');
                        writeln('Bilangan tersebut adalah',bil_acak, '. ');
                    end
                else
                    if tebakan > bil_acak then
                        begin
                            writeln('Aduh, kurang daripada ',tebakan);
                            writeln('(Anda masih punya ',kesempatan,'kesempatan menebak. ');
                        end
                    else
                        begin
                            writeln('Ah, lebih besar daripada ',tebakan);

```

```

        writeln('(Anda masih punya ',kesempatan,'kesempatan menebak.));
        end;
        end;
end;
writeln;
writeln('Anda ingin bermain lagi? (Y/T)');
readln(pilih); if upcase(pilih)='Y' then goto mulai;
end.
-----

```

2. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
program tblfgsi;
uses wincrt;
var
    x      : real;
    lagi   : char;
label hitung;
begin
    writeln('Menghitung nilai berbagai fungsi matematika. ');
    x:=-23.456; writeln('Contoh: x = ',x:5:3);
    writeln('Harga mutlak      : |x|      = ',abs(x), ' (REAL)');
    writeln('Bagian pecahan    : Frac(x) = ',frac(x), ' (REAL)');
    writeln('Bagian bulat      : Int(x)   = ',int(x), ' (REAL)');
    writeln('Logaritma alami    : Ln(|x|) = ',ln(abs(x)), ' (REAL)');
    writeln('Pembulatan          : Round(x)= ',round(x), '(LongInt)':24);
    writeln('Nilai bulat          : Trunc(x)= ', trunc(x), '(LongInt)':24);
    writeln('Ekspnensial          : e^x      = ', exp(x), ' (REAL)');
    writeln('Notasi mEp artinya m x 10^p. ');
    write('Tekan ENTER');readln;
    hitung:
    clrscr;
    write('Masukkan bil REAL x = '); readln(x);
    writeln('Harga mutlak      : |x|      = ',abs(x), ' (REAL)');
    writeln('Bagian pecahan    : Frac(x) = ',frac(x), ' (REAL)');
    writeln('Bagian bulat      : Int(x)   = ',int(x), ' (REAL)');
    writeln('Logaritma alami    : Ln(|x|) = ',ln(abs(x)), ' (REAL)');
    writeln('Pembulatan          : Round(x)= ',round(x), '(LongInt)':24);
    writeln('Nilai bulat          : Trunc(x)= ', trunc(x), '(LongInt)':24);
    writeln('Ekspnensial          : e^x      = ', exp(x), ' (REAL)');
    writeln('Notasi mEp artinya m x 10^p. ');
    writeln;write('Hitung lagi ? <Y/T>_');readln(lagi);
    if upcase(lagi)='Y' then goto hitung;
    donewincrt;
end.
-----

```

3. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
program trigonom;
uses wincrt;
var
    sudut, sudut1, sudut2, i, langkah: integer;
    radian : real;

begin
    clrscr;
    writeln('Tabel Trigonometri');
    write('Masukkan sudut terkecil : ');readln(sudut1);
    write('Masukkan sudut terbesar : ');readln(sudut2);
    for i:=1 to 50 do write('='); writeln;

```



```

        writeln('Sudut(' ,chr(176),') ' , ' Sudut
(Radian) ' , 'Cos':7, 'Sin':8, 'Tan':8);
        for i:=1 to 50 do write('-'); writeln;
        if (sudut2-sudut1)<=10 then langkah:=1
        else if (sudut2-sudut1)<=20 then langkah:=2
        else if (sudut2-sudut1)<=30 then langkah:=3
        else if (sudut2-sudut1)<=40 then langkah:=4
        else if (sudut2-sudut1)<=50 then langkah:=5
        else if (sudut2-sudut1)<=60 then langkah:=6
        else if (sudut2-sudut1)<=70 then langkah:=7
        else if (sudut2-sudut1)<=80 then langkah:=8
        else if (sudut2-sudut1)<=90 then langkah:=9
        else if (sudut2-sudut1)<=100 then langkah:=10
        else langkah:=20;
        sudut:=sudut1;
        while sudut<=sudut2 do
        begin
            radian:=sudut*Pi/180;
            write(sudut:6,#176,radian:12:2,cos(radian):13:2,sin(radian):8:2);
            if cos(radian)<>0 then writeln(sin(radian)/cos(radian):8:2)
            else writeln('Inf':8);
            sudut:=sudut+langkah;
        end;
        for i:=1 to 50 do write('='); writeln;
    end.
-----

```

4. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

Program Tripytha;
uses wincrt;
var  n,p,q,r,i : integer;

begin
    writeln('Mencetak daftar Triple Pythagoras ...');
    write('Maksimum nilai yang diinginkan, 2<n<182 : ');
    readln(n);
    writeln('Daftar Tripel Pythagorasnya adalah:');
    i:=1;
    for r:=3 to n do
        for q:=1 to r-1 do
            for p:=1 to q do
                if (p<>q) AND (p*p+q*q=r*r) then
                    begin
                        write(i:2,') . ' , ' (' ,p:4,' , ' ,q:4,' , ' ,r:4,' )');
                        writeln('      Cek: ' ,p*p:5,' + ':3,q*q:6,' = ':3,r*r:7);
                        inc(i);
                        if (i mod 21 = 0) then
                            begin
                                write('Tekan ENTER untuk melihat lanjutannya ...');
                                readln;
                            end;
                    end;
            end;
        end;
    end.
-----

```

5. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

Program Faktorbi;
uses wincrt;
var  a,p,bf:integer; {a bilangan asli, p pembagi, bf = cacah faktor}
    lagi:char;
-----

```

```

label baca;
begin
  baca: clrscr;
  Write('Masukkan sebuah bilangan asli > 1 : ');readln(a);
  if a<2 then goto baca;
  Writeln('Faktor-faktor dari ',a,' adalah: ');
  writeln(1:5);
  bf:=2;    {minimal cacah faktor}
  if a>3 then
    for p:=2 to round(int(a/2)) do if a mod p = 0 then
      begin
        writeln(p:5);
        inc(bf);      {berarti p adalah faktor a}
      end;
  writeln(a:5);
  if bf=2 then writeln('Jelas bahwa ', a,' adalah prima!')
  else writeln ('Jelas bahwa ', a,' adalah komposit!');
  write('Mau coba lagi? <Y/T>: ');readln(lagi);
  if upcase(lagi)='Y' then goto baca;
  donewincrt;
end.
-----

```

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

A. Judul Praktikum : **Praktikum Pemrograman Komputer**

B. Identitas Praktikan :

Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_

Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_

C. Hasil Praktikum :

1. **Compile** program-program yang Anda tulis tersebut. Cek apakah terdapat kesalahan. Jika masih salah, betulkan. Jika sudah benar, cek dengan masukan yang sesuai, apakah outputnya benar.
2. Setelah program Anda benar error, jalankan program-program Anda di atas beberapa kali dengan data yang berbeda-beda, kemudian tuliskan hasilnya pada bagian ini.
3. Tuliskan perintah-perintah Pascal baru yang baru saja Anda ketahui dan jelaskan kegunaannya.

# Praktikum 6 (Enumerasi dan Himpunan)

## A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

## B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menyebutkan jenis-jenis variabel dalam Turbo Pascal dan jangkauan nilai masing-masing jenis variabel.
4. Mahasiswa dapat membuat program Pascal untuk menampilkan elemen-elemen suatu daftar atau himpunan.
5. Mahasiswa dapat menggunakan perintah `type`, `const`, `ord`, `pred`, `succ`.
6. Mahasiswa dapat menulis program Pascal sederhana yang menggunakan jenis data enumerasi dan himpunan.
7. Mahasiswa dapat menulis program yang memerlukan masukan (input) untuk pengulangan program.

## C. Dasar Teori

Operator pada himpunan (**set**) meliputi:

Operator	Operasi	Jenis Data	jenis hasil
$\leq$	subhimpunan	set	Boolean
$\geq$	superhimpunan	set	Boolean
in	anggota	jenis ordinal (operand di kiri), set/himpunan yang terdiri tas jenis ordinal	Boolean
+	Union (Gabungan)	semua jenis set	
-	Selisih	semua jenis set	
*	Irisan	semua jenis set	

Hasil operasi himpunan mengikuti aturan logika himpunan:

- Suatu nilai ordinal C adalah anggota  $A + B$  hanya jika C adalah anggota A atau B.
- Suatu nilai ordinal C adalah anggota  $A - B$  hanya jika C adalah anggota A tetapi bukan anggota B.
- Suatu nilai ordinal C adalah anggota  $A * B$  hanya jika C adalah anggota A maupun B.

Apabila nilai ordinal terkecil anggota hasil suatu operasi himpunan adalah A dan nilai terbesarnya B, jenis hasil operasi tersebut adalah **set of A..B**.

Suatu deklarasi **type** mendefinisikan suatu pengenalan yang menyatakan suatu jenis data. Aturan penulisannya adalah sbb.:

```

type
  NamaJenis1 = Jenis;
  ...
  NamaJenisN = Jenis;

```

dengan Jenis adalah salah satu dari: **array, file, object, ordinal, pointer, real, record, set, string**

Turbo Pascal menyediakan enam kelas jenis pokok. Jenis sederhana mendefinisikan himpunan nilai-nilai yang terurut: Jenis **Ordinal**, Jenis **Integer**, Jenis **Boolean**, Jenis **Char**, Jenis **Enumerated**, Jenis **Subrange**, Jenis **Real**.

Turbo Pascal mendefinisikan sembilan jenis ordinal (dapat dibilang). Lima di antaranya termasuk jenis **integer** yang merupakan himpunan bagian khusus dari himpunan bilangan bulat. Keempat jenis ordinal lainnya adalah jenis boolean (**Boolean**, **WordBool**, **LongBool**), dan **Char**. Dua kelas jenis lain yang dapat digunakan untuk mendefinisikan jenis baru adalah jenis **enumerated** dan **subrange**.

Fungsi-fungsi baku yang dapat digunakan pada semua jenis ordinal adalah: **Ord, Pred, Succ**.

Jenis **boolean** dibagi menjadi tiga: **Boolean**, **WordBool**, dan **LongBool**.

### Sintaks:

```

type
  Boolean = (False, True);
  WordBool = (False, True);
  LongBool = (False, True);

```

### Catatan:

Ketiga jenis boolean tersebut masing-masing berukuran:

- jenis **Boolean** = satu-Byte (8 bits)
- jenis **WordBool** = seukuran Word (16 bits)
- jenis **LongBool** = seukuran Longint (32 bits)

Oleh karena jenis **boolean** termasuk jenis **ordinal enumerated**, maka berlaku:

```

False < True
Ord(False) = 0
Ord(True) = 1
Succ(False) = True
Pred(True) = False

```

**Boolean** merupakan jenis yang lebih disukai dan memakan lebih sedikit memori dibanding kedua jenis **boolean** lainnya. Keberadaan jenis **WordBool** dan **LongBool** untuk memberikan kecocokan (kompatibilitas) dengan lingkungan Windows.

Jenis **Enumerated** mendefinisikan suatu himpunan nilai-nilai terurut dengan memerikan nama-nama pengenalan yang menunjukkan nilai-nilai tersebut. Urutan nilai-nilainya sama dengan urutan penulisannya.

### Sintaks:

```

type
  NamaHimpunan = (nilai1, nilai2, ..., nilaiN );

```

### Catatan:

Pada definisi di atas, nilai1, nilai2, ..., nilaiN merupakan konstanta-konstanta untuk jenis NamaHimpunan. Konstanta nilai1 mempunyai nilai ordinal 0, nilai2 1, dan seterusnya,  $\text{ord}(\text{nilaiN}) = N - 1$ . Jenis **Enumerated** merupakan bagian dari jenis ordinal.

### Contoh:

```
type
  hari = (Ahad, Senin, Selasa, Rabu, Kamis, Jum'at, Sabtu);
```

Pada contoh di atas, Rabu adalah suatu nilai konstanta jenis hari.

```
Ord(Ahad)      = 0
Ord(Senin)     = 1
Ord(Selasa)    = 2
```

## D. Kegiatan Praktek

1. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----
Program enum1;
uses wincrt;
type
  namabulan =(Jan, Feb, Mar, April, Mei, Juni, Juli, Agust, Sept, Okt, Nop, Des);

var n, hari      :integer;
    bulan:namabulan;
    lagi : char;
const
  K: set of namabulan =[April..Sept];

begin
  randomize;
  writeln('Program Pemakaian Enumerasi/Daftar');
  repeat
  n:=random(12); bulan:=namabulan(n);
  write('Tahukah Anda bahwa bulan ');
  case bulan of
    Jan  : begin write(' Januari ');hari:=31;end;
    Feb  : begin write(' Pebruari ');hari:=28;end;
    Mar  : begin write(' Maret ');hari:=31;end;
    April: begin write(' April ');hari:=30;end;
    Mei  : begin write(' Mei ');hari:=31;end;
    Juni : begin write(' Juni ');hari:=30;end;
    Juli : begin write(' Juli ');hari:=31;end;
    Agust: begin write(' Agustus ');hari:=31;end;
    Sept : begin write(' September ');hari:=30;end;
    Okt  : begin write(' Oktober ');hari:=31;end;
    Nop  : begin write(' Nopember ');hari:=30;end;
    Des  : begin write(' Desember ');hari:=31;end;
  end;
  writeln('bulan ke-',ord(bulan)+1,' dan terdiri atas ',hari,' hari?');
  if (bulan in K) then
    writeln('Pada bulan tersebut biasanya musim kemarau.')
  else writeln('Pada bulan tersebut biasanya musim penghujan.');
```

```
writeln;
write('Mau tahu bulan lain? <Y/T>_');readln(lagi);
until upcase(lagi)<>'Y';
```

```
    donewincrt;  
end.
```

2. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----  
Program enum2;  
uses wincrt;  
type  
  namahari=(Ahad,Senin,Selasa,Rabu,Kamis,Jumat,Sabtu);  
var n      : integer;  
    hari   : namahari;  
    lagi   : char;  
const  
  Libur:set of namahari=[Ahad];  
begin  
  writeln('Program Pemakaian Enumerasi/Daftar');  
  repeat  
    randomize;  
    n:=random(7); hari:=namahari(n);  
    write('Jika sekarang hari ');  
    case hari of  
      Ahad   : write('Ahad ');  
      Senin  : write('Senin ');  
      Selasa : write('Selasa ');  
      Rabu   : write('Rabu ');  
      Kamis  : write('Kamis ');  
      Jumat  : write('Jum''at ');  
      Sabtu  : write('Sabtu ');  
    end;  
    if (hari in Libur) then  
      writeln('(Libur).')  
    else writeln('(Hari kerja).');  
    hari:=pred(namahari(n));  
    write('Maka hari sebelumnya adalah ');  
    case hari of  
      Ahad   : write('Ahad ');  
      Senin  : write('Senin ');  
      Selasa : write('Selasa ');  
      Rabu   : write('Rabu ');  
      Kamis  : write('Kamis ');  
      Jumat  : write('Jum''at ');  
      Sabtu  : write('Sabtu ');  
    else write('Sabtu ');  
    end;  
    if (hari in Libur) then  
      writeln('(Libur).')  
    else writeln('(Hari kerja).');  
    hari:=succ(namahari(n));  
    write('Hari sesudahnya adalah ');  
    case hari of  
      Ahad   : write('Ahad ');  
      Senin  : write('Senin ');
```

```

        Selasa : write('Selasa ');
        Rabu   : write('Rabu   ');
        Kamis  : write('Kamis  ');
        Jumat  : write('Jum' 'at ');
        Sabtu  : write('Sabtu  ');
        else write('Ahad  ');
    end;
    if (hari in Libur) then
        writeln('(Libur).')
        else writeln('(Hari kerja).');
    writeln;
    write('Mau tahu hari lain? <Y/T>_');readln(lagi);
    until (lagi in ['Y','y']) = false;
    donewincrt;
end.
-----

```

3. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program himpunan;
uses wincrt;
const
    HURUFBESAR:set of char =['A'..'Z'];
    hurufkecil:set of char =['a'..'z'];
    angka      :set of char =['0'..'9'];
    hurufhidup:set of char
        =['A','a','E','e','I','i','O','o','U','u'];

var k,huruf,lagi : char;

begin
    writeln('Program pengujian karakter input. ');
    repeat
        write('Ketikkan sebuah karakter : ');readln(huruf);
        if huruf in HURUFBESAR then
            begin
                write('Anda telah mengetik sebuah huruf besar ');
                if huruf in hurufhidup then writeln('(HURUF HIDUP).')
                else writeln('(HURUF MATI).');
            end
        else if huruf in hurufkecil then
            begin
                write('Anda telah mengetik sebuah huruf kecil ');
                if huruf in hurufhidup then writeln('(huruf hidup).')
                else writeln('(huruf mati).');
            end
        else if huruf in angka then
            writeln('Anda telah mengetik sebuah angka digit.')
            else writeln('Anda mengetik karakter khusus. ');
        writeln;
        write('Mau tahu hari lain? <Y/T>_');readln(lagi);
        until (lagi in ['Y','y']) = false;
    until (lagi in ['Y','y']) = false;
end.

```

```
    donewincrt ;  
end.
```

---

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

A. Judul Praktikum : **Praktikum Pemrograman Komputer**

B. Identitas Praktikan :

Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_

Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_

C. Hasil Praktikum :

1. **Compile** program-program yang Anda tulis tersebut. Cek apakah terdapat kesalahan. Jika masih salah, betulkan. Jika sudah benar, cek dengan masukan yang sesuai, apakah outputnya benar.
2. Setelah program Anda benar error, jalankan program-program Anda di atas beberapa kali dengan data yang berbeda-beda, kemudian tuliskan hasilnya pada bagian ini.
3. Tuliskan perintah-perintah Pascal baru yang baru saja Anda ketahui dan jelaskan kegunaannya.



# Praktikum 7

## (Array: Variabel Berdimensi / Jajaran Variabel)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menyebutkan jenis-jenis variabel dalam Turbo Pascal dan jangkauan nilai masing-masing jenis variabel.
4. Mahasiswa dapat membuat program Pascal untuk menyimpan dan menampilkan sekumpulan data sejenis dengan menggunakan variabel *array*.
5. Mahasiswa dapat menggunakan perintah *array* untuk mendefinisikan variabel berdimensi satu (vektor) dan variabel berdimensi dua (matriks).
6. Mahasiswa dapat menulis program Pascal sederhana yang menggunakan variabel-variabel *array* untuk pengolahan data numerik/string.
7. Mahasiswa dapat menulis program yang memerlukan masukan (input) beberapa data.

### C. Dasar Teori

Sebuah **array** adalah suatu kelompok atau sekumpulan kelompok seperti beberapa item yang terdefinisi. Contoh sebuah definisi array adalah:

```
skor : array[1..100] of real;
```

yang mendefinisikan sebuah array bernama *skor* yang dapat digunakan untuk menyimpan sampai dengan 100 nilai riil: *skor[1]*, *skor[2]*, ..., *skor[100]*.

Array sering digunakan untuk menghemat nama variabel, dengan cara mengelompokkan sejumlah variabel sejenis dan memberinya nama yang sama, nilai yang satu dibedakan dengan dengan nilai yang lain oleh indeks. Hal ini akan mempermudah pengolahan data yang disimpan dengan variabel berindeks tersebut. Dalam hal ini, indeks berfungsi sebagai perujuk ke nilai tertentu dari variabel yang bersangkutan. Dalam contoh di atas, misalnya, *skor[25]* dapat digunakan untuk merujuk skor ke-25, dst. Contoh yang lain, misalkan kita ingin menyimpan data suhu selama satu bulan (30 hari). Sudah tentu kita dapat menggunakan 30 nama variabel yang berlainan untuk menyimpan data suhu hari pertama sampai hari ke-30. Namun cara ini tidak efisien dan menyulitkan pengolahan data kita. Cara terbaik adalah dengan menggunakan array, misalnya:

```
suhu : array[1..30] of real;
```

Untuk memahami bagaimana proses perujukan ke nilai tertentu dari sebuah variabel berindeks, kita perlu mengetahui bagaimana data dalam suatu array disimpan di dalam memori komputer. Terdapat sebuah aturan bahwa: SEMUA DATA DISIMPAN DI DALAM MEMORI KOMPUTER SECARA LINIER. Hal itu berarti bahwa data yang satu disimpan pada lokasi yang berurutan dengan lokasi data sebelumnya. Itulah sebabnya kita dapat menggunakan indeks untuk merujuk nilai tertentu pada suatu array. Sebagai indeks perujuk dapat kita gunakan ekspresi matematis.

Jika kita ingin mencetak data suhu selama satu bulan baris demi baris, kita dapat menggunakan variabel berindeks di atas sebagai berikut:

```
for i := 1 to 30 do
    writeln('Suhu pada hari ke-',i,' : ',suhu[i]);
```

Array juga dapat digunakan pada loop **for/while/repeat** dengan menggunakan variabel indeks (dalam contoh di atas adalah **i** sebagai **penghitung**). Variabel penghitung harus berjenis ordinal, misalnya **integer** atau **char**, sesuai dengan nilai-nilai yang tercantum pada definisi array yang bersangkutan, seperti:

```
array1: array[1..10] of integer;
array2: array[-15..0] of real;
array3: array['a'..'z'] of char;
```

Kita dapat mendefinisikan, asalkan memori komputer mencukupi, sejumlah tak berhingga array. Sebagai contoh, misalkan kita ingin mendefinisikan sebuah array untuk menyimpan data suhu selama satu tahun (12 bulan, tiap bulan maksimum 31 hari). Dalam hal ini kita dapat menggunakan definisi:

```
suhu: array[1..12] of array[1..31] of integer;
```

atau

```
suhu: array[1..12, 1..31] of integer;
```

Di dalam memori komputer data suhu tersebut akan disimpan dengan urutan sebagai berikut:

```
suhu[1,1] suhu[1,2] ... suhu[1,31] suhu[2,1] .... suhu[12,31]
```

dan `suhu[i, j]` menyatakan suhu pada hari ke-*j* pada bulan ke-*i*.

Berikut adalah contoh kode untuk mencetak data suhu selama satu tahun pada layar dalam bentuk 12 baris:

```
for i := 1 to 12 do
    begin
        for j := 1 to 31 do
            write(suhu[i, j], ' ');
        writeln;
    end;
```

Sebuah array juga dapat digunakan untuk menyimpan beberapa konstan, seperti pada contoh berikut:

```
const
    bulan: array[1..12] of string[3] = ('Jan', 'Feb', 'Mar', 'Apr', 'Mei',
        'Jun', 'Jul', 'Agu', 'Sep', 'Okt', 'Nov', 'Des');
```

## D. Kegiatan Praktek

1. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program olahdata;
uses wincrt;
var x:array[1..100] of real; {maksimum 100 data}
    n,i,j:integer;
    s,m,v,d:real;
begin
writeln('Program Olah Data Sederhana:');writeln;
write('Cacah data: 100 <= ');readln(n);
writeln('Data acak dalam rentang 0 - 100:');
writeln;
{hasilkan data langsung hitung jumlahnya}
s:=0;
randomize;
for i:=1 to n do
    begin
        x[i]:=random(101);
        writeln(' Data ke-',i,': ',x[i]:5:2);
        s:=s+x[i];
        if i mod 15 = 0 then
            begin write('Tekan ENTER ...');readln;end;
        end;
m:=s/n; {rata-rata}
v:=0;{hitung varians}
for i:=1 to n do v:=v+sqr((x[i]-m));
v:=v/n;
{urutkan data dari terbesar ke terkecil ...}
for i:=1 to n-1 do
    for j:=i+1 to n do
        if x[i]<x[j] then {tukar nilai x[i] dan x[j]}
            begin
                d:=x[i];x[i]:=x[j];x[j]:=d;
            end;

    {cetak data ...}
writeln;
writeln('Jumlah data = ',s:5:3);
writeln('Rata-rata = ',m:5:3);
writeln('varians = ',v:5:3);
writeln('Simpangan baku = ',sqrt(v):5:3);
writeln;
writeln('Data setelah diurutkan:');writeln;
for i:=1 to n do
begin write(x[i]:8:2); if i mod 8 = 0 then writeln; end;
-----

```

2. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program Tripels;
uses wincrt;
var A,B,C : array[1..100] of integer;
    n : integer;
    p,q,r,i: integer;

```

```

begin
  writeln('Mencetak daftar Triple Pythagoras ...');
  write('Maksimum nilai yang diinginkan, 2<n<182 : ');
  readln(n);
  i:=1;
  for r:=3 to n do
    for q:=1 to r-1 do
      for p:=1 to q do
        if (p<>q) AND (p*p+q*q=r*r) then
          begin
            A[i]:=p;B[i]:=q;C[i]:=r;
            write(A[i]:5,B[i]:5,C[i]:5);
            writeln('    Cek: ':5,p*p:5,'+':3,q*q:6,'=':3,r*r:7);
            inc(i);
            if (i mod 21 = 0) then
              begin
                write('Tekan ENTER untuk melihat lanjutannya ...');
                readln;
              end;
          end;
        end;
      end;
    end;
  end.
-----

```

3. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program Pascal3;
uses wincrt;
var P : array[1..25,1..25] of integer;
    n : integer;
    i,j: integer;

begin
  writeln('Mencetak Segitiga Pascal ...');
  write('Banyaknya baris yang diinginkan, n<=15 : ');
  readln(n);
  P[1,1]:=1; writeln(P[1,1]:5);
  for i:=2 to n do
    begin
      P[i,1]:=1;write(P[i,1]:5);
      for j:=2 to i do
        begin
          P[i,j]:=P[i-1,j-1]+P[i-1,j]; write(P[i,j]:5);
        end;
      writeln;
    end;
  end.
-----

```

4. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program Fakbils;
uses wincrt;
type pembagi= array[1..50] of integer;

```

```

var a:integer;
    faktor:pembagi;
    lagi:char;
    p,bf:integer;
label baca;
begin
    baca:
    clrscr;
    Write('Masukkan sebuah bilangan asli > 1 : ');
    readln(a); if a<2 then goto baca;
    bf:=1;faktor[bf]:=1;
    if a>3 then
        for p:=2 to round(int(a/2)) do if a mod p = 0 then
            begin inc(bf); faktor[bf]:=p; end;
    inc(bf);faktor[bf]:=a;
    {cetak faktor;}
    Writeln('Faktor-faktor dari ',a,' adalah: ');
    for p:=1 to bf do writeln(faktor[p]:5);
    if bf=2 then writeln('Jelas bahwa ', a,' adalah prima!')
        else writeln ('Jelas bahwa ', a,' adalah komposit!');
    write('Mau coba lagi? <Y/T>: ');readln(lagi);
    if upcase(lagi)='Y' then goto baca;
    donewincrt;
end.
-----

```

5. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program BSA3;
uses wincrt;
var BSA : array[1..15,1..15] of integer;
    n,i,j,k,jumlah: integer;
label baca;
begin
    baca:
    writeln('Bujur Sangkar Ajaib Berukuran Ganjil');
    write('Banyaknya baris yang diinginkan, n<=15 : ');
    readln(n);
    if (n mod 2 = 0) or (n<3) or (n>15) then goto baca;
    i:=1;j:=(n+1) div 2;
    for k:=1 to n*n do
        begin
            BSA[i,j]:=k;
            if k mod n = 0 then
                if i+1>n then i:=1 else inc(i);
            if k mod n <>0 then
                begin
                    if i-1<1 then i:=n else dec(i);
                    if j+1>n then j:=1 else inc(j);
                end;
        end;
    jumlah:=0;
    for i:=1 to n do

```

```

begin
jumlah:=jumlah+BSA[i,i];
for j:=1 to n do write(BSA[i,j]:5);
writeln;
end;
writeln('Jumlah baris/kolom/diagonal = ',jumlah);
end.
-----

```

6. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program Fibonacci;
uses wincrt;
var F : array[1..100] of integer;
    n,i: integer;
label baca;
begin
baca:
writeln('Barisan Bilangan Fibonacci');
write('Banyaknya suku barisan, n<=23 : ');
readln(n);
if (n<3) or (n>23) then goto baca;
F[1]:=1;F[2]:=1;
for i:=3 to n do F[i]:=F[i-1]+F[i-2];
for i:=1 to n do
begin
write(F[i]:6);
if i mod 10 = 0 then writeln;
end;
end.
-----

```

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

A. Judul Praktikum : **Praktikum Pemrograman Komputer**

B. Identitas Praktikan :

Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_

Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_

C. Hasil Praktikum :

1. Jalankan program-program Anda di atas beberapa kali dengan data yang berbeda-beda, kemudian tuliskan hasilnya pada bagian ini.
2. Tuliskan perintah-perintah Pascal baru yang baru saja Anda ketahui dan jelaskan kegunaannya.

# Praktikum 8

## (Variabel String: Operasi dan Fungsi-fungsi String)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menyebutkan jenis-jenis variabel dalam Turbo Pascal dan jangkauan nilai masing-masing jenis variabel.
4. Mahasiswa dapat membuat program Pascal untuk melakukan manipulasi data teks / string.
5. Mahasiswa dapat menggunakan perintah `chr`, `upcase`, `ord`, `delete`, `pos`, dan `copy` untuk melakukan operasi dan manipulasi string.
6. Mahasiswa dapat menulis program Pascal sederhana yang menggunakan operasi-operasi karakter dan string.

### C. Dasar Teori

Berikut adalah daftar fungsi/prosedur string dan karakter pada Turbo Pascal.

Nama Fungsi/ Prosedur	Pemakaian	Hasil
Chr	Chr(X: Byte): Char;	Karakter dengan kode ASCII X
Concat	Concat(s1 [, s2,..., sn]: String): String;	Gabungan string s1, s2, ...,sn
Copy	Copy(S: String; P: Integer; N: Integer): String;	Mengambil karakter pada string S mulai dari posisi P sebanyak N karakter ke belakang
Length	Length(S: String): Integer;	Cacah karakter pada string S
Pos	function Pos(Substr: String; S: String): Byte;	Posisi ditemukannya (karakter pertama) Substr pada string S
Pred	Pred(X): < Sejenis dengan X >;	Karakter pendahulu (sebelum) X
Succ	Succ(X): <Sejenis dengan X >;	Karakter sesudah X
UpCase	UpCase(K: Char): Char;	Mengubah karakter K menjadi huruf besar
Delete	Delete(var S: String; Index: Integer; Count:Integer);	Menghapus string <b>S</b> dari karakter ke <b>Index</b> sebanyak <b>Count</b> karakter.
Str	Str(X [: Width [: Decimals ]]; var S);	Menkonversi bilangan X menjadi string S.
Val	Val(S; var V; var Kode: Integer);	Mengkonversi string (angka) S menjadi nilai numerik
Insert	Insert(Sisip: String; var S: String; Index: Integer);	Menyisipkan string <b>Sisip</b> ke string <b>S</b> pada posisi ke <b>Index</b> .

## Operator String

Di dalam Turbo Pascal, operator + dapat digunakan untuk menggabungkan dua buah nilai berjenis string atau Char, menghasilkan nilai jenis string. Apabila panjang string hasil penggabungan lebih dari 255 karakter, maka akan dipotong menjadi 255 karakter.

## D. Kegiatan Praktek

1. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----  
Program string1;  
uses wincrt;  
type  
    teks=string[10];  
var  
    teks1,teks2 : teks;  
    n1,n2,i,j   : integer;  
    lagi       : char;  
label baca;  
begin  
    baca: clrscr;  
    writeln('Pengolahan String (Teks)');  
    write('Tuliskan teks I, maksimum 10 karakter: ');  
    readln(teks1);  
    write('Tuliskan teks II, maksimum 10 karakter: ');  
    readln(teks2);  
    writeln;  
    writeln('Teks pertama Anda adalah: ',teks1);  
    n1:=length(teks1);  
    write('Terdiri atas ',n1);  
    writeln(' karakter (termasuk spasi).');  
    write('Jika dibalik tulisan tersebut menjadi: ');  
    for i:=n1 downto 1 do write(teks1[i]);writeln;  
    write('Jika ditulis dengan HURUF BESAR: ');  
    for i:=1 to n1 do write(upcase(teks1[i]));writeln;  
    writeln;  
    writeln('Teks kedua Anda adalah: ',teks2);  
    n2:=length(teks2);  
    write('Terdiri atas ',n2);  
    writeln(' karakter (termasuk spasi).');  
    write('Jika dibalik tulisan tersebut menjadi: ');  
    for i:=n2 downto 1 do write(teks2[i]);writeln;  
    write('Jika ditulis dengan HURUF BESAR: ');  
    for i:=1 to n2 do write(upcase(teks2[i]));writeln;  
    writeln;  
    writeln('Urutan kedua teks tersebut adalah:');  
    if teks1 <= teks2 then writeln(teks1,', ',teks2)  
        else writeln(teks2,', ',teks1);  
    writeln('Tekan ENTER ...');readln;  
    writeln('Huruf-huruf yang sama pada kedua teks:');  
    for i:=1 to n1 do  
        for j:=1 to n2 do  
            if upcase(teks1[i])=upcase(teks2[j]) then
```



```

        begin
        write(teks1[i], '(posisi ke-', i, ') dan ');
        write(teks2[j], '(posisi ke-', j, ').');
        writeln
        end;
    write('Coba lagi?<Y/T>_'); readln(lagi);
    if upcase(lagi)='Y' then goto baca;
    donewincrt;
end.
-----

```

2. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program string2;
uses wincrt;
type
    teks=string[50];
var
    kalimat, sisa : teks;
    n, i, ls : integer;
    lagi : char;
    suku, temp : string[5];
    posisi : array[1..50] of integer;
    jumlah : integer;

label baca;
begin
    baca: clrscr;
    writeln('Pencarian suku kata dalam kalimat');
    writeln('Tuliskan sebuah kalimat, maksimum 50 karakter: ');
    readln(kalimat);
    write('Tuliskan sebuah suku-kata: '); readln(suku);
    writeln;
    ls:=length(suku); jumlah:=0;
    sisa[0]:=kalimat[0];
    for i:=1 to length(kalimat) do sisa[i]:=upcase(kalimat[i]);
    for i:=1 to ls do
temp[i]:=upcase(suku[i]); temp[0]:=suku[0];

        n:=pos(temp, sisa);
        while n<>0 do
        begin
            inc(jumlah); posisi[jumlah]:=n+ls*(jumlah-1);
            delete(sisa, n, ls);
            n:=pos(temp, sisa);
        end;
    writeln('Kalimat: ', kalimat);
    writeln('memuat ', jumlah, ' suku kata ', suku);
    if jumlah>0 then
    begin
        write('Yakni pada posisi ke-');
        for i:=1 to jumlah do write(posisi[i]:3); writeln;
    end;
end.

```

```
writeln;
write('Coba lagi?<Y/T>_');readln(lagi);
if upcase(lagi)='Y' then goto baca;
donewincrt;
end.
```

3. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----
Program Kar2bin;
uses wincrt;
type
    kodebiner=string[8];
var
    karakter,lagi : char;
    kode          : kodebiner;
    ascii        : byte;
    i            : integer;
label mulai;
begin
    mulai: clrscr;
    writeln('Kode ASCII dan BINER 8-bit suatu karakter');
    write('Ketikkan sebuah karakter : ');readln(karakter);
    ascii:=ord(karakter);
    kode:='';
    for i:=1 to 8 do
    begin
        if (ascii AND 1)=1 then kode:='1'+kode
        else kode:='0'+kode;
        ascii:=ascii SHR 1;
    end;
    write('Karakter ',karakter,' mempunyai nilai ASCII #');
    writeln(ord(karakter),' dan kode biner 8-bit ',kode);
    write('Coba lagi?<Y/T>_');readln(lagi);
    if upcase(lagi)='Y' then goto mulai;
    donewincrt;
end.
```

4. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----
Program string3;
uses wincrt;
type
    teks=string[100];
var
    kalimat      : teks;
    n,i          : integer;
    lagi,huruf: char;

label baca;
begin
    baca: clrscr;
```

```

writeln('Mengubah huruf besar menjadi huruf kecil');
writeln('Tuliskan kalimat, maks 100 karakter, huruf besar
semua: ');
readln(kalimat);
write('Jika ditulis dengan huruf kecil menjadi: ');
writeln;
n:=length(kalimat);
for i:=1 to n do
begin
huruf:=kalimat[i];
if (huruf>='A') AND (huruf<='Z') then
begin
huruf:=chr(ord(huruf)+32);
kalimat[i]:=huruf;
end;
end;
writeln(kalimat);
writeln;
write('Coba lagi?<Y/T>_');readln(lagi);
if upcase(lagi)='Y' then goto baca;
donewincrt;
end.

```

5. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program string4;
uses wincrt;
type
teks=string[100];
var
kalimat,hasil : teks;
k,n,p : integer;
lagi : char;

label baca;
begin
baca: clrscr;
writeln('Mengkopi String dari String (Kumpulan Karakter)');
writeln('Tuliskan sebuah kalimat, maksimum 100 karakter:
');
readln(kalimat);
write('Ingin mengkopi mulai karakter ke-'); readln(k);
write('Cacah huruf yang hendak dikopi : ');readln(p);
writeln;

n:=length(kalimat); if p>n then p:=n-k;
hasil:=copy(kalimat,k,p);
writeln('Kalimat yang Anda
masukkan:');writeln(#34,kalimat,#34);
writeln(p,' karakter dari posisi ke-',k,' adalah
',#34,hasil,#34);
writeln;

```

```
write('Coba lagi?<Y/T>_');readln(lagi);
if upcase(lagi)='Y' then goto baca;
donewincrt;
end.
```

-----

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

A. Judul Praktikum : **Praktikum Pemrograman Komputer**

B. Identitas Praktikan :

Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_

Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_

C. Hasil Praktikum:

1. **Compile** program-program yang Anda tulis tersebut. Cek apakah terdapat kesalahan. Jika masih salah, betulkan. Jika sudah benar, cek dengan masukan yang sesuai, apakah outputnya benar.
2. Setelah program Anda benar error, jalankan program-program Anda di atas beberapa kali dengan data yang berbeda-beda, kemudian tuliskan hasilnya pada bagian ini.
3. Tuliskan perintah-perintah Pascal baru yang baru saja Anda ketahui dan jelaskan kegunaannya.

# Praktikum 9

## (Pemrograman Prosedural: Prosedur, Fungsi, dan Rekursi)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menyebutkan jenis-jenis variabel dalam Turbo Pascal dan jangkauan nilai masing-masing jenis variabel.
4. Mahasiswa dapat membuat program Pascal yang menggunakan prosedur dan fungsi.
5. Mahasiswa dapat membedakan dan menggunakan perintah `procedure` dan `function` secara tepat.
6. Mahasiswa dapat menulis program Pascal yang menggunakan prosedur rekursif untuk melakukan suatu proses tertentu.
7. Mahasiswa dapat menulis program Pascal yang menggunakan fungsi rekursif untuk menghitung suatu fungsi matematika.

### C. Dasar Teori

Program yang rumit dapat ditulis secara sistematis dengan mengelompokkan beberapa perintah tertentu menjadi satu kesatuan sebagai suatu prosedur atau fungsi. Hal ini sangat bermanfaat selain untuk menyederhanakan program juga untuk menghemat penulisan perintah. Baik prosedur maupun fungsi harus didefinisikan pada bagian deklarasi suatu program Pascal.

Suatu prosedur adalah bagian dari suatu program yang digunakan untuk melakukan suatu tugas tertentu, mungkin dengan menggunakan beberapa parameter tertentu. Format pendefinisian suatu prosedur adalah sbb:

```
procedure NamaProsedur;  
begin  
    statemen1;  
    statemen2;  
    ...  
end;  
atau  
procedure  
NamaProsedur(parameter);  
begin  
    statemen1;  
    statemen2;  
    ...  
end;
```

Kata **procedure** digunakan untuk mendefinisi nama prosedur dan parameter (jika ada) yang dapat digunakan pada saat memanggil prosedur tersebut di dalam bagian utama program Pascal.

Bagian utama suatu prosedur diawali dengan kata **Begin** dan diakhir dengan kata **End;** (diikuti tanda titik koma!) Antara kata **Begin** dan **End;** dapat dideklarasikan beberapa variabel, konstanta, dll. sebagaimana pada program Pascal. Deklarasi ini bersifat lokal, hanya berlaku dalam prosedur yang bersangkutan, tidak dapat digunakan di dalam program utama. Dengan kata lain, struktur suatu prosedur mirip dengan dengan struktur program Pascal itu sendiri. Statemen-statemen antara **Begin** dan **End;** akan dikerjakan apabila nama prosedur tersebut dipanggil di dalam bagian utama program Pascal.

Paramater di dalam suatu prosedur dapat digunakan sebagai pengganti bagian deklarasi lokal dalam prosedur tersebut.

### Contoh:

```
{ Mendefinisikan suatu prosedur }
procedure TulisStr(X, Y: integer; S: string);
var
  PosisiX, PosisiY: Integer;
begin
  PosisiX := WhereX;
  PosisiY := WhereY;
  GotoXY(X, Y);
  Write(S);
  GotoXY(PosisiX, PosisiY);
end;
```

Suatu fungsi di dalam program Pascal adalah bagian program yang berfungsi untuk melakukan perhitungan dan menghasilkan suatu nilai. Format pendefinisian suatu fungsi adalah sbb:

```
function NamaFungsi: jenis;
begin
  statemen1;
  statemen2;
  ...
end;
                                atau
function
  NamaFungsi(parameter): jenis;
begin
  statemen1;
  statemen2;
  ...
```

Kata **function** digunakan untuk mendefinisikan nama fungsi, parameter (jika ada) dan jenis nilai hasil perhitungan. Struktur suatu fungsi mirip dengan struktur suatu prosedur, bedanya fungsi harus menghasilkan suatu nilai. Bagian utama suatu fungsi harus memuat sedikitnya sebuah statemen yang memberi nilai pada NamaFungsi. Hasil dari fungsi yang bersangkutan adalah nilai terakhir NamaFungsi.

### Contoh:

```
(* Definisi suatu fungsi *)
function StrHurufBesar(S: string): string;
var
  I: Integer;
  for I := 1 to Length(S) do
  if (S[I] >= 'a') and (S[I] <= 'z') then
  Dec(S[I], 32);
  StrHurufBesar := S; {Nilai fungsi}
StrHurufBesar}
end;
```

Fungsi atau prosedur yang di dalamnya memanggil dirinya sendiri (fungsi atau prosedur itu) disebut **fungsi rekursif** atau **prosedur rekursif**.

## D. Kegiatan Praktek

1. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----
program prosedul;
uses wincrt;
var
  x, y      : integer;
```

```

    hasil      : integer;
    hasilkali: longint;
    lagi       : char;

procedure tambah(bil1, bil2: integer; var jumlah: integer);
begin
    jumlah := bil1 + bil2;
end;

procedure selisih(bil1, bil2: integer; var sisa: integer);
begin
    sisa := bil1 - bil2;
end;

procedure kali(bil1, bil2: longint; var hasil: longint);
begin
    hasil := bil1*bil2;
end;

label baca;
begin
    baca: clrscr;
    writeln('Operasi dua buah bilangan bulat dengan prosedur');
    writeln('Masukkan dua buah bilangan (pisahkan dengan spasi).');
    readln(x, y);
    tambah(x, y, hasil);
    writeln('Jumlahnya adalah ', hasil, '.');
    selisih(x, y, hasil);
    writeln('Selisihnya adalah ', hasil, '.');
    kali(x, y, hasilkali);
    writeln('Hasil kalinya adalah ', hasilkali, '.');
    write('Coba lagi?<Y/T>_');readln(lagi);
    if upcase(lagi)='Y' then goto baca;
    donewincrt;
end.
-----

```

2. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
program fungsil;
uses wincrt;
var
    x, y : integer;
    lagi : char;

function tambah(bil1, bil2: integer): integer;
begin
    tambah := bil1 + bil2;
end;

function selisih(bil1, bil2: integer): integer;
begin
    selisih := bil1 - bil2;
end;

function kali(bil1, bil2: longint): longint;
begin
    kali := bil1*bil2;
end;

label baca;
begin

```

```

baca: clrscr;
writeln('Operasi dua buah bilangan bulat dengan fungsi:');
writeln('Masukkan dua buah bilangan (pisahkan dengan spasi).');
readln(x, y);
writeln('Jumlahnya adalah ', tambah(x,y), '.');
writeln('Selisihnya adalah ', selisih(x,y), '.');
writeln('Hasil kalinya adalah ', kali(x,y), '.');
write('Coba lagi?<Y/T>_');readln(lagi);
if upcase(lagi)='Y' then goto baca;
donewincrt;
end.

```

3. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program Permtasi;
uses wincrt;
type
  permutasi=array[1..10] of char;
var
  P      : permutasi;
  i, K   : byte;
  lagi   : char;
  total  : longint;

procedure masukan(N:byte;var P:permutasi);
var
  i      : byte;
  huruf  : char;
begin
  for i:=1 to N do
  begin
    write('Karakter ke-',i,' : '); readln(huruf);
    P[i]:=huruf;
  end;
end;

procedure CetakPerm(P: permutasi;awal:byte;var total:longint);
var
  i      : byte;
  tmp    : char;
begin
  if awal=K then
  begin
    for i:=1 to K do write(P[i]);
    writeln;inc(total);
    if (total mod 20 = 0) then
    begin writeln('Tekan ENTER ...');readln; end;
  end
  else
  begin
    for i:=awal to K do
    begin
      tmp:=P[i];P[i]:=P[awal];P[awal]:=tmp;
      CetakPerm(P,awal+1,total);
    end;
  end;
end;
label mulai;
begin
  mulai: clrscr;
  writeln('Program Permutasi Huruf');
  write('Masukkan cacah huruf<=10 : ');readln(K);

```



```

if (K<1) or (K>10) then goto mulai;
masukan(K,P);
write('Permutasi lengkap dari huruf ');
for i:=1 to K do write(P[i],', '); writeln(' adalah :');
total:=0;
CetakPerm(P,1,total);
writeln('Terdapat sebanyak ', total,' permutasi.');
```

write('Mau coba lagi? <Y/T>: ');readln(lagi);

```

if upcase(lagi)='Y' then goto mulai;
donewincrt;
end.
```

- 
4. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).
- 

```

Program Matriks3;
uses wincrt;
type matrix3 = array[1..3,1..3]of integer; {definisi matriks 3 x 3}
var a,b,c,d,e,h:matrix3;

procedure baca_matriks(var a:matrix3); {untuk memasukkan matriks 3 x 3}
var i:integer;
begin
    {baca matriks 3 x 3 baris demi baris }
    for i:=1 to 3 do readln(a[i,1],a[i,2],a[i,3]);
end;

procedure cetak_matriks(a:matrix3); {mencetak matriks 3 x 3}
var i: integer;
begin
    {cetak matriks 3 x 3 baris demi baris}
    for i:=1 to 3 do writeln(a[i,1]:4,a[i,2]:4,a[i,3]:4);
end;

procedure jumlah(a,b:matrix3;var m:matrix3); {hitung m=a+b}
var i,j: integer;
begin
    for I:=1 to 3 do
        for j:=1 to 3 do
            m[i,j]:=a[i,j]+b[i,j];
end;

procedure selisih(a,b:matrix3;var m:matrix3); {hitung m=a-b}
var i,j: integer;
begin
    for I:=1 to 3 do
        for j:=1 to 3do
            m[i,j]:=a[i,j]-b[i,j];
end;

procedure kali(a,b:matrix3; var m:matrix3); {hitung m=ab}
var i,j,k: integer;
begin
    for I:=1 to 3 do
        for j:=1 to 3 do
            begin
                {hitung c[i,j] = a[i,1]*[b[1,j] +
                ...+a[i,3]*b[3,j]}
                m[i,j]:=0;
                for k:=1 to 3 do
                    m[i,j]:=m[i,j]+a[i,k]*b[k,j];
            end;
end;

procedure transpose(a:matrix3; var m:matrix3); {hitung m=a'}
var i,j:integer;
```

```

begin
  for i:=1 to 3 do
    for j:=1 to 3 do
      m[i,j]:=a[j,i];
    end;
  end;

begin {bagian utama}
writeln('Program Operasi Matriks 3 x 3');
writeln('Caca memasukkan matriks:');
writeln('Antar elemen sebaris pisahkan dengan SPASI. ');
writeln('Antar baris pisahkan dengan ENTER. ');
writeln('Masukkan matriks A:'); baca_matriks(a);
writeln('Masukkan matriks B:'); baca_matriks(b);
jumlah(a,b,c); selisih(a,b,d); kali(a,b,e); transpose(a,h);
write('Tekan ENTER ...'); readln;
writeln('A = '); cetak_matriks(a);
writeln('B = '); cetak_matriks(b);
writeln('A + B = '); cetak_matriks(c);
writeln('A - B = '); cetak_matriks(d);
writeln('AxB = '); cetak_matriks(e);
writeln('A'' = '); cetak_matriks(h);
end.

```

5. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program rekursil;
uses wincrt;

function faktorial(n:byte):longint;
begin
  if (n=0) or (n=1) then faktorial:=1
  else faktorial:=n*faktorial(n-1);
end;

function Fibonacci(n:byte):longint;
begin
  if (n=0) then Fibonacci:=0 else
  if (n=1) then Fibonacci:=1 else
  Fibonacci:=Fibonacci(n-1)+Fibonacci(n-2);
end;

function FPB(m,n:integer):integer;
begin
  if (m=0) then FPB:= n else
  if (m<n) then FPB:=FPB(n,m)
  else FPB:=FPB(m mod n,n);
end;

function KPK(m,n:integer):integer;
begin
  KPK:=m*n div FPB(m,n);
end;

function kombinasi(m,k:integer):integer;
begin
  if m<k then kombinasi:= 0 else
  kombinasi:=faktorial(m) div (faktorial(k)*faktorial(m-k));
end;

var
  pilih,lagi: char;
  i,n       : byte;

```

```

    a,b      : integer;

label mulai;

begin
    mulai: clrscr;
    writeln('Program Matematika:');
    writeln('=====');
    writeln('1. Barisan Fibonacci');
    writeln('2. Menghitung Faktorial');
    writeln('3. Menghitung FPB dan KPK');
    writeln('4. Menghitung Nilai Kombinasi');
    writeln('5. Keluar Program');
    write('Masukkan nomor pilihan Anda :_');readln(pilih);
    case pilih of
    '1': begin
            write('Masukkan cacah suku : ');readln(n);
            for i:=1 to (n-1) do write(Fibonacci(i),' ');
            writeln(Fibonacci(n));
        end;
    '2': begin
            write('Masukkan sebuah bilangan asli : ');readln(n);
            writeln(n,'! = ',faktorial(n));
        end;
    '3': begin
            write('Masukkan dua buah bilangan asli : ');
            readln(a,b);
            writeln('FPB(',a,',',b,') = ',FPB(a,b));
            writeln('KPK(',a,',',b,') = ',KPK(a,b));
        end;
    '4': begin
            write('Masukkan dua buah bilangan asli : ');
            readln(a,b);
            writeln('C(',a,',',b,') = ',kombinasi(a,b));
        end;
    '5': donewincrt;

    else writeln('Anda salah memasukkan pilihan!');
    end;
    write('Mau coba lagi? <Y/T>: ');readln(lagi);
    if upcase(lagi)='Y' then goto mulai;
    donewincrt;
end.
-----

```

6. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program Menu1;
uses wincrt;
Var a,b,n,pilihan:integer;
    lagi:char;
label baca,awal;

Procedure Faktor(a:integer);
var x,bf:integer;
    pembagi:array[1..50] of integer;
begin
    pembagi[1]:=1;bf:=1;    { 1 adalah faktor a}
    if a>3 then {kita perlu menghitung semua faktor a}
        for x:=2 to a-1 do if a mod x = 0 then
            begin
                bf:=bf+1;    {berarti x adalah faktor a}
            end;
    end;
end;
-----

```

```

                pembagi[bf]:=x; {simpan nilai x}
            end;
            bf:=bf+1;pembagi[bf]:=a; { a merupakan faktor a}
        {cetak faktor:}
        Writeln('Faktor-faktor dari ',a,' adalah: ');
        for x:=1 to bf do writeln(pembagi[x]:5);
        if bf=2 then writeln('Jelas bahwa ', a,' adalah prima!')
        else writeln ('Jelas bahwa ', a,' adalah komposit!');
    end;

Procedure Fibonacci(a,b,n:integer);
var Fib:array[1..50] of integer;
    i:integer;
begin
    Fib[1]:=a; Fib[2]:=b;i:=2;
    while i<n do
        begin
            i:=i+1;
            Fib[i]:=Fib[i-2]+Fib[i-1];
        end;
    for i:=1 to n do
        begin
            write(Fib[i]);if i<n then write(' ');
        end;
        writeln;
    end;

Procedure Menu;
begin
    clrscr;
    writeln('Program Memilih Menu');
    writeln('=====');writeln;
    writeln('Menu Pilihan:');writeln;
    writeln('1. Faktor Bilangan');writeln;
    writeln('2. Barisan Fibonacci');writeln;
    writeln('3. Keluar');writeln;

end;

begin {Program utama}
    awal:
    Menu; {tampilkan menu}
    repeat
        write('Tulis nomor pilihan Anda: ');readln(pilihan);
    until pilihan in [1,2,3];
    {case pilihan of}

    if pilihan=1 then
        begin
            repeat
                baca: Write('Masukkan sebuah bilangan asli > 1 : ');
                readln(a);
                if a<2 then goto baca;
                Faktor(a);
                write('Mau coba lagi? <Y/T>: ');readln(lagi);
            until upcase(lagi)<>'Y';
            goto awal;
        end;

    if pilihan=2 then
        begin
            repeat
                Write('Masukkan suku ke-1, 2, dan banyak suku (n) : ');
                readln(a,b,n);

```

```

        Fibonacci(a,b,n);
        write('Mau coba lagi? <Y/T>: ');readln(lagi);
        until upcase(lagi)<>'Y';
        goto awal;
    end;
    if pilihan=3 then writeln('Terima kasih!');
end.
-----

```

7. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
program dec2base; { mengubah basis bilangan }
uses wincrt;

function pangkat(x, y: integer):longint;
{menghitung x pangkat y, diperlukan pada fungsi desimal }
var
    i, hasil: longint;
begin
    hasil := 1;
    for i := 1 to y do
        hasil := hasil * x;
    pangkat := hasil;
end;

function nondesimal(x: longint; basis: integer):string;
{ mengubah basis 10 ke basis basis lain < 37 }
const
    simbolbil: string = '0123456789ABCDEFGHIJKLMNPOQRSTUVWXYZ';
    { simbol-simbol untuk lambang bilangan }
var
    i: integer;
    hasil, hasilbagi, sisa: longint;
    simpan: string;
begin
    if basis > 36 then
        nondesimal := '!' {karakter penanda kegagalan fungsi ini }
    else
        begin
            simpan := '';
            hasil := x;
            while hasilbagi <> 0 do
                begin
                    hasilbagi := hasil div basis;
                    sisa := hasil mod basis;
                    simpan := simbolbil[sisa + 1] + simpan;
                    { penyajian sesungguhnya }
                    hasil := hasilbagi;
                end;
            nondesimal := simpan;
        end;
    end;
end;

function desimal(x: string; basis: integer):longint;
{ mengubah sembarang basis < 37 ke basis 10 }
const
    simbolbil: string = '0123456789ABCDEFGHIJKLMNPOQRSTUVWXYZ';
var
    i, pkt: integer;
    hasil: longint;
    cacahsimbol: integer;
begin
    if basis > 36 then

```

```

        desimal := -1 {untuk menandai kegagalan fungsi ini }
    else
    begin
        hasil := 0;
        i := 1;
        pkt := length(x) - 1;
        while i <> length(x)+1 do    { kembalikan ke basis 10 }
            begin
                hasil := hasil + (pos(x[i], simbolbil)-1) *
                    pangkat(basis, pkt);
                i := i + 1;
                pkt := pkt - 1;
            end;
        desimal := hasil;
    end;

var
    bil, basis1: longint;
    basis2: string;
    lagi : char;
label mulai;
begin
    mulai: clrscr;
    writeln('Program konversi basis bilangan:');
    writeln('-----');
    write('Tuliskan sebuah bilangan (basis 10): ');
    readln(bil);
    write('Ingin diubah ke dalam basis berapa? ');
    readln(basis1);
    writeln;
    basis2 := nondesimal(bil, basis1);
    if basis2 = '!' then
        writeln('Tuliskan basis yang kurang dari 37')
    else begin
        write(bil, ' (basis 10) sama dengan ', basis2);
        writeln(' (basis ',basis1, ').');
        bil := 0;
        write('Untuk mengecek: ',basis2, ' (basis ', basis1);
        writeln(') sama dengan ',desimal(basis2, basis1),' (basis 10).');
    end;
    write('Mau coba lagi?<Y/T>_ ');readln(lagi);
    if upcase(lagi)='Y' then goto mulai;
    donewincrt;
end.
-----

```

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

- A. Judul Praktikum : **Praktikum Pemrograman Komputer**
- B. Identitas Praktikan :
- |                      |                       |
|----------------------|-----------------------|
| Nama Mhs : _____     | NIM : _____           |
| Praktikum ke : _____ | Tgl Praktikum : _____ |
- C. Hasil Praktikum :

1. **Compile** program-program yang Anda tulis tersebut. Cek apakah terdapat kesalahan. Jika masih salah, betulkan. Jika sudah benar, cek dengan masukan yang sesuai, apakah outputnya benar.
2. Setelah program Anda benar error, jalankan program-program Anda di atas beberapa kali dengan data yang berbeda-beda, kemudian tuliskan hasilnya pada bagian ini.
3. Tuliskan perintah-perintah Pascal baru yang baru saja Anda ketahui dan jelaskan kegunaannya.

# Praktikum 10

## (Bekerja dengan RECORD)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menyebutkan jenis-jenis variabel dalam Turbo Pascal dan jangkauan nilai masing-masing jenis variabel.
4. Mahasiswa dapat menggunakan perintah `type` untuk mendefinisikan jenis data `record`.
5. Mahasiswa dapat membedakan dan menggunakan perintah `read/readln` dan `write/writeln` secara tepat untuk membaca dan menulis data berjenis `record`.
6. Mahasiswa dapat menulis program Pascal sederhana yang menggunakan operasi-operasi data `record`
7. Mahasiswa dapat menulis program Pascal sederhana yang menggunakan variabel array berjenis `record`.

### C. Dasar Teori

Dalam Pascal, kita dapat mengelompokkan beberapa jenis data yang berbeda dengan menggunakan `record`. Untuk mendefinisikan sebuah `record`, kita harus menggunakan perintah **type** dan diakhiri dengan **end;**. Berikut adalah contoh ddefinisi sebuah `record` untuk menyimpan data mahasiswa, yang terdiri atas nama, nomor mahasiswa, program studi, dan alamat:

```
type
  mahasiswa = record
    nama: string[25]
    nim: longint;
    prodi: string[20];
    alamat: string[50];
  end;
```

Dalam definisi di atas, kita mengelompokkan empat buah variabel (`nama`, `nim`, `prodi`, `alamat`) ke dalam sebuah jenis `record` bernama `mahasiswa`. Dalam hal ini, `mahasiswa` adalah jenis variabel baru yang memiliki empat komponen. Selanjutnya kita dapat mendefinisikan variabel baru berjenis `mahasiswa`.

Kita dapat bekerja dengan semua atau sebagian variabel yang merupakan bagian dari sebuah **record**. Keseluruhan `record` dapat diakses dengan memanggil variabel `record` yang digunakan. Untuk mengakses variabel yang merupakan bagian dari suatu `record` digunakan nama `record` diikuti nama variabel yang bersangkutan dan dipisahkan dengan tanda titik (`.`). Perintah **WITH** dapat digunakan untuk mempermudah pengetikan dan mengurangi kemungkinan salah ketik karena kurang tanda titik di dalam bekerja dengan sebuah `record`.



Pada array variabel record, indeks diletakkan sesudah nama variabel record, BUKAN setelah nama variabel yang merupakan bagian dari record.

#### D. Kegiatan Praktek

1. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----  
program record1;  
uses wincrt;  
type  
  mahasiswa = record  
    nama: string[20];  
    nim: string[10];  
    prodi: string[20];  
    alamat: string[30];  
  end;  
var  
  data_mhs: array[1..50] of mahasiswa;  
  n,i: integer;  
begin  
  clrscr;  
  writeln('Contoh pemakaian record: Data Mahasiswa');  
  write('Masukkan cacah data: ');readln(n);  
  for i:=1 to n do  
  begin  
    writeln('data ke-',i);  
    write('nama mahasiswa: ');readln(data_mhs[i].nama);  
    write('nomor mahasiswa: ');readln(data_mhs[i].nim);  
    write('program studi: ');readln(data_mhs[i].prodi);  
    write('alamat: ');readln(data_mhs[i].alamat);  
  end; {perhatikan pemakaian indeks untuk array !}  
  writeln;  
  write('Tekan ENTER ...');readln;  
  clrscr;  
  writeln('Tabel Data');  
  for i:=1 to n do  
  begin  
    with data_mhs[i] do    {#32 = SPASI}  
    begin  
      write(i,#32,nama,#32:20-length(nama));  
      write(nim,#32:8-length(nim));  
      write(prodi,#32:20-length(prodi));  
      writeln(alamat);  
    end;  
  end;  
end.  
-----
```

2. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----  
program Kalimat1;
```

```

uses
  WinCrt;
const
  PanjangBaris = 128;
  LebarKata    = 10;

type
  Kalimat      = String[PanjangBaris];
  DataHuruf    = record
    Cacah      : Word;
    AwalKata   : Word;
  end;

var
  Baris, Kata  : Word;
  Huruf        : LongInt;
  TabeLebarKata: array[1..LebarKata] of Word;
  TabelHuruf   : array['A'..'Z'] of DataHuruf;
  Masukan      : Kalimat;
  i            : byte;

procedure InfoDataHuruf(Awal, Akhir: Char);
var
  S: Char;
begin
  Writeln; Write('Huruf      :');
  for S := Awal to Akhir do Write(S:5);
  Writeln; Write('Frekuensi  :');
  for S := Awal to Akhir do Write(TabelHuruf[S].Cacah:5);
  Writeln; Write('Di awal kata:');
  for S := Awal to Akhir do Write(TabelHuruf[S].AwalKata:5);
  Writeln;
end;

procedure CetakHasil;
var
  i      : Integer;
  KataPerbaris: Real;
begin
  if Baris <> 0 then KataPerbaris := Kata / Baris
  else KataPerbaris := 0;
  Writeln;
  Writeln('Anda telah menuliskan ',Baris, ' baris (kalimat) ');
  writeln('yang memuat ',Kata, ' kata dan ',Huruf,' huruf. ');
  Writeln('Rata-rata kata per baris: ',KataPerbaris:0:2, '. ');
  Writeln;
  Write('Huruf/Kata: ');
  for i := 1 to LebarKata do Write(i:4);
  Writeln;

  Write('Frekuensi : ');
  for i := 1 to LebarKata do Write(TabeLebarKata[i]:4);
  Writeln;
  InfoDataHuruf('A', 'M');
  InfoDataHuruf('N', 'Z');
end;

procedure Inisialisasi;
begin
  Baris := 0;
  Kata  := 0;
  Huruf := 0;
  FillChar(TabelHuruf, SizeOf(TabelHuruf), 0);

```

```

    FillChar(TabeLebarKata, SizeOf(TabeLebarKata), 0);
end;

function CekHuruf(S: Char): Boolean;
begin
    CekHuruf := UpCase(S) in ['A'..'Z'];
end;

procedure CekBaris(var S: Kalimat);
var
    i      : Integer;
    LebarHuruf: Word;

begin
    Inc(Baris);
    i := 1;
    while i <= Length(S) do
    begin
        while (i <= Length(S)) and not CekHuruf(S[i]) do Inc(i);
        LebarHuruf := 0;
        while (i <= Length(S)) and CekHuruf(S[i]) do
        begin
            Inc(Huruf);
            Inc(TabelHuruf[UpCase(S[i])].Cacah);
            if LebarHuruf = 0 then Inc(TabelHuruf[UpCase(S[i])].AwalKata);
            Inc(i);
            Inc(LebarHuruf);
        end;
        if LebarHuruf > 0 then
        begin
            Inc(Kata);
            if LebarHuruf <= LebarKata then
                Inc(TabeLebarKata[LebarHuruf]);
        end;
    end;
end;

function BacaBaris(i:byte): Kalimat;
var
    S: Kalimat;

begin
    Write('Kalimat ke-',i,' : ');
    Readln(S); BacaBaris := S;
end;

begin
    Writeln('Program Membaca kalimat/string dan memprosesnya. ');
    Writeln('Tuliskan sebuah kalimat & baris kosong untuk mengakhiri. ');
    Inisialisasi;
    i:=1;
    Masukan := BacaBaris(i);inc(i);
    while Masukan <> '' do
    begin
        CekBaris(Masukan);
        Masukan := BacaBaris(i);inc(i);
    end;
    CetakHasil;
end.
-----

```

3. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).
-

```

Program Matrixmn;
uses wincrt;
const maxbaris=10;
      maxkolom=10;
type matrix = record
  baris,kolom : byte;
  elemen      : array[1..maxbaris,1..maxkolom] of real;
end;

procedure baca_matriks(var A:matrix;m,n:byte);
var i,j:byte;
begin
  A.baris:=m; A.kolom:=n;
  for i:=1 to m do
    for j:=1 to n do
      begin
        write('Elemen ke-(',i,',',j,') = ');
        readln(A.elemen[i,j]);
      end;
    end;
end;

procedure cetak_matriks(A:matrix);
var i,j: byte;
begin
  for i:=1 to A.baris do
    begin
      for j:=1 to A.kolom do write(A.elemen[i,j]:7:2);
      writeln;
    end;
end;

procedure jumlah(A,B:matrix;var M:matrix);
var i,j: byte;
begin
  if (A.baris <> B.baris) or (A.kolom <>B.kolom) then
    writeln('Error, matriks tidak cocok!') else
    begin
      M.baris:=A.baris;M.kolom:=A.kolom;
      for i:=1 to A.baris do
        for j:=1 to A.kolom do
          M.elemen[i,j]:=A.elemen[i,j]+B.elemen[i,j];
        end;
      end;
end;

procedure selisih(A,B:matrix;var M:matrix);
var i,j: byte;
begin
  if (A.baris <> B.baris) or (A.kolom<>B.kolom) then
    writeln('Error, matriks tidak cocok!') else
    begin
      M.baris:=A.baris;M.kolom:=A.kolom;
      for i:=1 to A.baris do
        for j:=1 to A.kolom do
          M.elemen[i,j]:=A.elemen[i,j]-B.elemen[i,j];
        end;
      end;
end;

procedure kali(A,B:matrix;var M:matrix);
var i,j,k: byte;
begin
  if A.kolom<>B.baris then
    writeln('Error, matriks tidak cocok!') else
    begin
      M.baris:=A.baris;

```

```

M.kolom:=B.kolom;
for i:=1 to M.baris do
  for j:=1 to M.kolom do
    begin
      M.elemen[i,j]:=0;
      for k:=1 to A.kolom do
        M.elemen[i,j]:=M.elemen[i,j]+A.elemen[i,k]*B.elemen[k,j];
      end;
    end;
  end;
end;

procedure transpose(A:matrix; var M:matrix);
var i,j:byte;
begin
  M.baris:=A.kolom;
  M.kolom:=A.baris;
  for i:=1 to M.baris do
    for j:=1 to M.kolom do
      M.elemen[i,j]:=A.elemen[j,i];
    end;
  end;

var A,B,C,D,E,H:matrix;
    m,n: byte;

begin {bagian utama}
  writeln('Program Operasi Matriks');
  writeln('Maksimum ukuran matriks 10 x 10. ');
  writeln('Menentukan ukuran matriks. ');
  writeln('Matriks A: ');
  write('Cacah baris = '); readln(m);
  write('Cacah kolom = '); readln(n);
  baca_matriks(A,m,n);
  writeln('Matriks B: ');
  write('Cacah baris = '); readln(m);
  write('Cacah kolom = '); readln(n);
  baca_matriks(B,m,n);
  write('Tekan ENTER ... '); readln;
  writeln('A = '); cetak_matriks(A);
  writeln('B = '); cetak_matriks(B);
  write('Tekan ENTER ... '); readln;
  writeln('A + B = '); jumlah(A,B,C); cetak_matriks(C);
  writeln('A - B = '); selisih(A,B,D); cetak_matriks(D);
  write('Tekan ENTER ... '); readln;
  writeln('AxB = '); kali(A,B,E); cetak_matriks(E);
  writeln('A' = '); transpose(A,H); cetak_matriks(H);
  write('Tekan ENTER untuk keluar ... '); readln;
  donewincrt;
end.

```

4. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
Program Hanoi1;
uses WinCrt;

const maxkeping=10;
type
  menara=record
    nama   : char;
    cacah  : byte;
    keping: array[1..maxkeping] of byte;
  end;

```

```

procedure cetakmenara(A:menara);
var i: byte;
begin
    write(A.nama, '[');
    for i:=1 to A.cacah do write(A.keping[i]:3);
    writeln(']');
end;

procedure pindahkeping(var A,B:menara);
begin
    writeln(A.keping[A.cacah], '[', A.nama, ' --> ', B.nama, ']');
    inc(B.cacah);
    B.keping[B.cacah]:=A.keping[A.cacah];
    dec(A.cacah);
    write('Tekan ENTER ...');readln;
end;

procedure hanoi(var A,B,C:menara;n:byte;var langkah:longint);
begin
    if n=1 then
        begin
            cetakmenara(A);cetakmenara(B);cetakmenara(C);
            inc(langkah);write('Langkah ke-',langkah,': ');
            pindahkeping(A,C);
            writeln('Selesai!');Exit;
        end
    else hanoi(A,C,B,n-1,langkah);
        cetakmenara(A);cetakmenara(B);cetakmenara(C);
        inc(langkah);write('Langkah ke-',langkah,': ');
        pindahkeping(A,C);
        hanoi(B,A,C,n-1,langkah);
end;

var i,n      : byte;
    langkah: longint;
    A,B,C    : menara;
    lagi     : char;
label mulai;
begin
    mulai: clrscr;
    writeln('Program Menara Hanoi:');
    writeln('Memindahkan n keping batu dari menara A ke C');
    writeln('dengan perantara menara B. ');
    writeln('Keping diberi nomor n (terbesar), (n-1), ..., 2, 1. ');
    writeln('Nomor besar harus di bawah nomor kecil. ');
    writeln('=====');
    write('Tuliskan cacah keping, n <= 10 : ');readln(n);
    if (n<1) or (n>10) then goto mulai;
    A.nama:='A';B.nama:='B';C.nama:='C';
    A.cacah:=n; B.cacah:=0;C.cacah:=0;
    for i:=1 to n do A.keping[i]:=(n+1-i);
    langkah:=0;
    hanoi(A,B,C,n,langkah);
    cetakmenara(A);cetakmenara(B);cetakmenara(C);
    write('Coba lagi? <Y/T>_ ');readln(lagi);
    if upcase(lagi)='Y' then goto mulai;
    donewincrt;
end.
-----

```

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

A. Judul Praktikum : **Praktikum Pemrograman Komputer**

B. Identitas Praktikan :

Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_

Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_

C. Hasil Praktikum :

1. **Compile** program-program yang Anda tulis tersebut. Cek apakah terdapat kesalahan. Jika masih salah, betulkan. Jika sudah benar, cek dengan masukan yang sesuai, apakah outputnya benar.
2. Setelah program Anda benar error, jalankan program-program Anda di atas beberapa kali dengan data yang berbeda-beda, kemudian tuliskan hasilnya pada bagian ini.
3. Tuliskan perintah-perintah Pascal baru yang baru saja Anda ketahui dan jelaskan kegunaannya.

# Praktikum 11

## (Bekerja dengan File: Membaca dan Menulis Data dari / ke File)

### A. Peralatan yang diperlukan:

1. Komputer (PC)
2. Program Pascal (Turbo Pascal for Windows, TPW 1.5)

### B. Tujuan Praktikum

1. Mahasiswa dapat menulis dan membetulkan kesalahan program Pascal.
2. Mahasiswa dapat menjalankan dan memberikan masukan program Pascal.
3. Mahasiswa dapat menyebutkan jenis-jenis variabel dalam Turbo Pascal dan jangkauan nilai masing-masing jenis variabel.
4. Mahasiswa dapat membuat program Pascal untuk membaca data yang tersimpan di dalam sebuah file dan menyimpan data ke dalam file teks.
5. Mahasiswa dapat membedakan dan menggunakan perintah `assign`, `rewrite`, `reset`, `eol`, `eof`, `close`, `write` dan `writeln` secara tepat untuk mengakses file teks.
6. Mahasiswa dapat menulis program Pascal sederhana yang memerlukan data masukan dari sebuah file.
7. Mahasiswa dapat menulis program Pascal sederhana yang menyimpan data output ke dalam sebuah file.

### C. Dasar Teori

Berikut adalah daftar fungsi/prosedur (perintah) Pascal yang berkaitan dengan operasi berkas.

Prosedur	Deklarasi	Kegunaan
Append	Append( File); File merupakan variabel berjenis <b>text</b> .	Membuka file untuk menambah isinya.
Assign	Assign(var F; File);	Menyimpan nama File ke variabel file F. (Tidak membuka File itu sendiri.)
ChDir	ChDir(S: String);	Pindah direktori (folder).
Close	Close(var F);	Menutup file yang namanya tersimpan dalam variabel file F
Erase	Erase(var F);	Menghapus file yang namanya tersimpan dalam variabel file F.
GetDir	GetDir(D: Byte; var S: String); dengan D bernilai 0= Default, 1=Drive A, 2=Drive B, 3=Drive C	Mengetahui direktori aktif
MkDir	MkDir(S: String);	Membuat direktori (folder) S.
Read	Read( [ var F: Text; ] V1 [, V2, ..., Vn ] );	Membaca data (yang dipisahkan dengan spasi) pada file yang namanya tersimpan dalam variabel file F ke variabel V1, V2, ...



Readln	Readln([ var F: Text; ] V1 [, V2, ..., Vn ]);	Membaca data di dalam file F baris demi baris.
Rename	Rename(var F; NamaBaru);	Mengganati nama file F dengan <b>NamaBaru</b> .
Reset	Reset(var F [: File; Recsize: Word ] );	Membuka file F.
Rewrite	Rewrite(var F: File [; Recsize: Word ] );	Membuat dan membuka file baru F.
Rmdir	Rmdir(S: String);	Menghapus direktori kosong S..
Write	Write( [ var F: Text; ] V1 [,V2,...,Vn ] );	Menyimpan nilai variabel V1, V2, ... Vn ke file F
Writeln	Writeln([ var F: Text; ] V1 [, V2, ..., Vn ] );	Menyimpan nilai variabel V1, V2, ... Vn ke file F dalam satu baris
Eof	Eof [ (var F: Text) ]: Boolean;	Status akhir file (end-of-file)
Eoln	Eoln [(var F: Text) ]: Boolean;	Status akhir baris (end-of-line).
FilePos	FilePos(var F): Longint;	Posisi dalam suatu file.
FileSize	FileSize(var F): Longint;	Ukuran suatu file.

Untuk mengakses sebuah file, kita dapat menggunakan statemen **assign**. Untuk merujuk suatu file digunakan sebuah variabel teks, sedangkan variabel string digunakan untuk merujuk **nama\_path** dan **nama file** tersebut. Untuk membuka/membaca sebuah file digunakan statemen **reset(variabel\_file)**; Untuk menuliskan sebuah file barus digunakan statemen **rewrite(variabel\_file)**; Statemen **close(variabel\_file)**; digunakan untuk menutup sebuah file setelah selesai diakses (dibaca/ditulis).

Terdapat beberapa fungsi yang didefinisikan di dalam Pascal baku yang digunakan di dalam pengolahan file-file teks. Fungsi-fungsi tersebut adalah **eof(variabel\_file)** dan **eoln(variabel\_file)**. Fungsi-fungsi ini menghasilkan nilai boolean dan merupakan signal yang dapat digunakan di dalam loop untuk menandai posisi di dalam file-file teks. Signal **EOF** menandai akhir sebuah file. Signal **EOLN** menandai akhir sebuah baris teks di dalam suatu file.

#### D. Kegiatan Praktek

1. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```
-----
program file1;
uses wincrt;
var
  infile, outfile      : text;
  inputchar            : char;
  berkas1,berkas2: string;
begin
  writeln('Program Baca dan Tulis Berkas Teks');
  write('Tulis nama berkas yang hendak dibaca: ');
  readln(berkas1);
  writeln('Semua huruf akan diubah ke huruf besar ...');
  write('Tulis nama berkas baru setelah diubah: ');
```

```

readln(berkas2);
assign(infile,berkas1);
reset(infile);
assign(outfile,berkas2);
rewrite(outfile);
while not eof(infile) do
begin
    while not eoln(infile) do
    begin
        read(infile, inputchar);
        write(outfile, upcase(inputchar));
    end;
    writeln(outfile);
    readln(infile);
end;
close(infile);
close(outfile);
writeln('Selesai ... silakan cek berkas ',berkas2);
end.

```

- 
2. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).
- 

```

program file2;
uses wincrt;
var
    infile,outfile : text;
    inputstring    : string;
    berkas1,berkas2: string;
function upstr(instring: string):string;
var
    i: integer;
    newstr: string;
begin
    newstr := '';
    for i := 1 to length(instring) do
        newstr := newstr + upcase(instring[i]);
    upstr:=copy(newstr,1,length(newstr));
end;

begin
    writeln('Program Baca dan Tulis Berkas Teks');
    write('Tulis nama berkas yang hendak dibaca: ');
    readln(berkas1);
    writeln('Semua huruf akan diubah ke huruf besar ...');
    write('Tulis nama berkas baru setelah diubah: ');
    readln(berkas2);
    assign(infile,berkas1);
    reset(infile);
    assign(outfile,berkas2);
    rewrite(outfile);
    readln(infile, inputstring);
    while not eof(infile) do
    begin
        writeln(outfile, upstr(inputstring));
        readln(infile, inputstring);
    end;
    writeln(outfile, upstr(inputstring));
    close(infile);

```

```

    close(outfile);
    writeln('Selesai ... silakan cek berkas ',berkas2);
end.

```

3. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

program file3;
uses wincrt;
type
    banyak_huruf = array['A'..'Z'] of longint;
var
    cacah          : banyak_huruf;
    namafile1,namafile2: string;
    total_huruf    : longint;
    simpan         : char;

procedure hitung_huruf(baris: string; var cacah: banyak_huruf);
var
    i: integer;
    k: char;
begin
    for i := 1 to length(baris) do
        for k := 'A' to 'Z' do
            if upcase(baris[i]) = k then
                begin
                    inc(cacah[k]);
                    total_huruf:= total_huruf + 1;
                end;
        end;
    end;

procedure displaydata(var cacah: banyak_huruf);
var
    i: integer;
    j: char;
begin
    writeln('Data hasil Pembacaan Teks');
    writeln('Pemakaian huruf pada berkas yang dibaca:');
    writeln('=====');
    writeln('Huruf':6,'Dipakai':10,'Huruf':12,'Dipakai':10);
    writeln('-----');
    for i := 1 to 13 do
        writeln(chr(i+64):4, cacah[chr(i+64)]:8,
                chr(i+77):14, cacah[chr(i+77)]:10);
    writeln('-----');
    writeln('Total banyaknya huruf yang dipakai : ',total_huruf);
end;

procedure baca_file(namafile: string; var cacah: banyak_huruf);
var
    infile: text;
    baris: string;
    k: char;
begin
    for k := 'A' to 'Z' do cacah[k]:=0;
    total_huruf:=0;
    assign(infile,namafile);
    reset(infile);
    readln(infile,baris );
    while not eof(infile) do
        begin
            hitung_huruf(baris,cacah);
            readln(infile, baris);
        end;
    end;
end;

```

```

        end;
        hitung_huruf(baris,cacah);
        close(infile);
    end;

procedure tulis_file(namafile: string; var cacah: banyak_huruf);
var
    file_simpan: text;
        i: integer;
        j: char;
begin
    assign(file_simpan,namafile);rewrite(file_simpan);
    writeln(file_simpan, 'Data hasil Pembacaan Teks');
    writeln(file_simpan, 'Nama file: ', namafile);
    writeln(file_simpan, 'Pemakaian huruf pada berkas yang dibaca:');
    writeln(file_simpan, '=====');
    writeln(file_simpan, 'Huruf':6, 'Dipakai':10, 'Huruf':12, 'Dipakai':10);
    writeln(file_simpan, '-----');
    for i := 1 to 13 do
        writeln(file_simpan, chr(i+64):4, cacah[chr(i+64)]:8,
                chr(i+77):14, cacah[chr(i+77)]:10);
    writeln(file_simpan, '-----');
    writeln(file_simpan, 'Total banyaknya huruf yang dipakai : ',
                total_huruf);

    close(file_simpan);
end;

begin {program utama}
    clrscr;
    writeln('Baca File Teks dan Hitung Huruf');writeln;
    write('Tulis nama file teks masukan : ');readln(namafile1);
    baca_file(namafile1,cacah); writeln;
    displaydata(cacah);
    write('Mau menyimpan data tersebut?<Y/T>_ ');readln(simpan);
    if upcase(simpan)='Y' then
        begin
            write('Hasilnya disimpan dengan nama : ');readln(namafile2);
            tulis_file(namafile2,cacah); writeln;
            writeln('Data file teks ',namafile1,
                    ' disimpan ke dalam file ',namafile2);

            end;
        donewincrt;
    end.
-----

```

4. Tulis program di bawah ini (persis apa adanya) dan simpan dengan nama sama dengan yang tertulis pada baris pertama (di belakang kata **program**).

```

-----
program file4;
uses wincrt,windos;
const
    pajak= 0.15;    {tingkat pajak penghasilan }
    bulan : array [0..11] of String[5] =
        ('Jan', 'Feb', 'Maret', 'April', 'Mei', 'Juni',
         'Juli', 'Agust', 'Sept', 'Okt', 'Nop', 'Des');
type
    string20 = string[20];
var
    karyawan,berkas1,berkas2: string20;
    jam_kerja, gaji_perjam: real;
    infile, outfile      : text;
    gaji_kotor, gaji_lembur, pph, gaji_bersih: real;
    y, bl, d, dow : Word;

```

```

function baris(karakter: char; panjang: integer):string;
var
    i: integer;
    str: string;
begin
    str := '';
    for i := 1 to panjang do str := str + karakter;
    baris := str;
end;

procedure tuliskop;    {tuliskan judul laporan/daftar gaji }
begin
    GetDate(y,bl,d,dow);
    writeln('PT. Barokah Makmur':45);
    writeln('Daftar Gaji Karyawan Bulan '+bulan[bl]+' ':50,y);
    writeln(baris('=' , 79));
    writeln('Nama', 'Jam kerja':21, 'Gaji/Jam':11,
            'Gaji kotor':12, 'Lembur':8, 'Pajak':8,
            'Gaji bersih':15);
    writeln(baris('-', 79));
end;

procedure simpan_judul;    {simpan judul laporan ke file}
begin
    GetDate(y,bl,d,dow);
    writeln(outfile,'PT. Barokah Makmur':45);
    writeln(outfile,'Daftar Gaji Karyawan Bulan '+bulan[bl]+' ':50,y);
    writeln(outfile,baris('=' , 79));
    writeln(outfile,'Nama','Jam kerja':21,'Gaji/jam':11,
            'Gaji kotor':12,'Lembur':8,'Pajak':8,
            'Gaji bersih':15);
    writeln(outfile,baris('-', 79));
end;

procedure hitung_gaji(jam, upah: real;var gkotor, tambahan,
                    potong, gbersih: real);
var
    gnormal: real;
begin
    if jam > 40 then {jika ada kelebihan jam kerja, maka}
    begin
        gnormal := upah * 40; {gaji normal sampai 40 jam kerja}
        tambahan := (jam - 40) * upah * 1.5; {gaji lembur}
    end
    else {tidak ada kelebihan jam kerja}
    begin
        gnormal := jam * upah;
        tambahan := 0;
    end;
    gkotor := gnormal + tambahan; {total gaji kotor}
    potong := gkotor * pajak;    {pph}
    gbersih := gkotor - potong;  {gaji bersih}
end;

begin
    writeln('Program Hitung Gaji Karyawan');
    writeln('PT. Barokah Makmur');
    writeln(baris('=' ,18));
    write('Tuliskan file masukan (data jam kerja): ');
    readln(berkas1);                {file masukan }
    assign(infile,berkas1);
    reset(infile);

```

```

write('Tuliskan nama file laporan (output): ');
readln(berkas2);          {file keluaran}
if berkas2<>' ' then
begin
  assign(outfile,berkas2);
  rewrite(outfile);
  simpan_judul;
end;
clrscr; tuliskop;
readln(infile, karyawan, jam_kerja, gaji_perjam);
while not eof(infile) do { baca file masukan sampai habis }
begin
  hitung_gaji(jam_kerja,gaji_perjam,gaji_kotor,gaji lembur,
             pph, gaji_bersih);
  writeln(karyawan,jam_kerja:5:2,gaji_perjam:10:1,
         gaji_kotor:11:1, gaji_lembur:10:1, pph:10:1,
         gaji_bersih:13:1);
  if berkas2<>' ' then
  writeln(outfile,karyawan, jam_kerja:5:2,gaji_perjam:10:1,
         gaji_kotor:11:1,gaji_lembur:10:1,pph:10:1,
         gaji_bersih:13:1);
  readln(infile, karyawan, jam_kerja, gaji_perjam);
end;
hitung_gaji(jam_kerja,gaji_perjam,gaji_kotor,gaji_lembur,
            pph, gaji_bersih);
writeln(karyawan, jam_kerja:5:2,gaji_perjam:10:1,
        gaji_kotor:11:1,gaji_lembur:10:1,pph:10:1,
        gaji_bersih:13:1);
writeln(baris('=', 79));
writeln('TTd. '#13'Bendahara');
if berkas2<>' ' then
begin
  writeln(outfile,karyawan, jam_kerja:5:2,gaji_perjam:10:1,
        gaji_kotor:11:1,gaji_lembur:10:1,pph:10:1,
        gaji_bersih:13:1);
  writeln(outfile,baris('=', 79));
  writeln(outfile,'TTd. '#13'Bendahara');
end;

close(infile);
close(outfile);    {tutup semua file}
end.

```

---

## E. Laporan Praktikum

Tulis laporan praktikum Anda dengan format sebagai berikut.

B. Judul Praktikum : **Praktikum Pemrograman Komputer**

C. Identitas Praktikan :

Nama Mhs : \_\_\_\_\_ NIM : \_\_\_\_\_

Praktikum ke : \_\_\_\_\_ Tgl Praktikum : \_\_\_\_\_

D. Hasil Praktikum :

1. **Compile** program-program yang Anda tulis tersebut. Cek apakah terdapat kesalahan. Jika masih salah, betulkan. Jika sudah benar, cek dengan masukan yang sesuai, apakah outputnya benar.

2. Setelah program Anda benar error, jalankan program-program Anda di atas beberapa kali dengan data yang berbeda-beda, kemudian tuliskan hasilnya pada bagian ini.
3. Tuliskan perintah-perintah Pascal baru yang baru saja Anda ketahui dan jelaskan kegunaannya.

## Daftar Pustaka

Abdul Kadir (2001). *Pemrograman Pascal, Buku 1*. Yogyakarta, Penerbit Andi.

Antony Pratama (2000). *Algoritma dan Pemrograman*. Yogyakarta, J& J Learning.

Rawlins, Grgory J.E. (1992) *Compare to What? An Introduction to the Analysis of Algorithms*. New York: W.H. Freeman and Company.

Sahid (1999). *Tutorial Pascal Mata Kuliah Pemrograman Komputer – Pengantar Bahasa Pemrograman Pascal dengan Turbo Pascal*. Yogyakarta, Jurusan Pendidikan Matematika FMIPA UNY.